GPU-Accelerated Mathematical Illustration

An Introduction to Shader Programming

Steve Trettel

January 2026

Table of contents

About

This mini-course introduces shader programming as a tool for mathematical illustration and exploration. Shaders are programs that run in parallel on the GPU, making them exceptionally fast for visualization tasks. We'll learn to write code that "reads like mathematics" using Shadertoy, a beginner-friendly web-based platform that handles all the low-level programming complexities.

We'll progress from 2D foundations (curves, tilings, fractals) to 3D rendering via raymarching. Along the way, we will implement classic examples like the Mandelbrot set, hyperbolic tessellations, and implicit surface renderers. The final day will explore either advanced geometric techniques (domain operations, 3D fractals) or temporal simulation methods (PDEs, buffer-based dynamics), depending on the group's interests.

No prior experience with shaders or GLSL is required—only a strong foundation in undergraduate mathematics and willingness to work hard and experiment with code through daily homework exercises. Here are some examples of things we will make:

1. Day 1: Introduction

1.1. Overview

By the end of today, you'll be able to create things like this:



A family of elliptic curves $y^2 = x^3 + ax + b$, drawn for several values of a simultaneously, with b varying across the screen. The curves shift in brightness to show the family structure, and you can watch singularities appear and disappear along the discriminant locus.

This image is computed in real time, every pixel evaluated independently on the GPU. To get here, we'll learn:

- What a shader is: a function from coordinates to colors, evaluated in parallel
- How to set up a coordinate system for mathematical visualization
- How to draw shapes using distance functions
- How to render implicit curves $F(x, y) = 0$ with uniform thickness
- How to add interactivity with mouse input

Let's begin.

1.2. What is a Shader?

We want to draw images on a screen.

Mathematically, an image is a function from a region $S \subset \mathbb{R}^2$ to the space of visible colors $\mathscr{C}$. This color space is three-dimensional, spanned by the responses of the three types of cone cells in our eyes. A convenient basis, roughly aligned with these responses, is red, green, and blue.

To realize this on a computer, we discretize. A screen is a grid of *pixels*: X pixels wide, Y pixels tall. Each pixel is a point in the integer lattice

$$\{0, 1, \ldots, X-1\} \times \{0, 1, \ldots, Y-1\}.$$

Colors are represented as RGB triples: red, green, and blue intensities, each in $[0, 1]$. The constraint to $[0, 1]$ reflects physical reality—a pixel has a maximum brightness it can display. (We can't draw the sun.) So an image is a function

$$f\colon \{0, \ldots, X-1\} \times \{0, \ldots, Y-1\} \to [0, 1]^3$$

$$(i, j) \mapsto (r, g, b).$$

In practice, we add a fourth component: *alpha*, representing transparency. This matters when compositing multiple layers (we won't use it in this course, but the machinery expects it). So our shader computes

$$f\colon (i, j) \mapsto (r, g, b, 1).$$

This is what a shader is. You write a function that takes pixel coordinates and returns an RGBA color. The GPU evaluates your function at every pixel to produce the image.

1.2.1. Parallelism

A 1920×1080 display has over two million pixels. How do we evaluate f at all of them fast enough to animate at 60 frames per second?

The answer is parallelism. A GPU contains thousands of cores, and it evaluates f at all pixels *simultaneously*. There's no loop over pixels in your code—you write f, and the hardware handles the rest.

The tradeoff: each pixel's computation must be *independent*. Pixel $(100, 200)$ cannot ask what color pixel $(100, 199)$ received. Every pixel sees the same global inputs—coordinates, time, mouse position—and must determine its color from those alone. Learning to think within this constraint is what shader programming is about.

> **Why "shader"?**
>
> The name comes from 3D graphics, where these programs computed *shading*—how light interacts with surfaces. It stuck even though we now use shaders for fractals, simulations, and mathematical visualization.

1.2.2. Why Shadertoy?

Shader programming normally requires substantial setup: OpenGL contexts, buffer management, compilation, render loops. Shadertoy abstracts all of this—you write one function, press play, and see results. We'll use it throughout the course.

1.3. First Shaders: Colors and Syntax

1.3.1. The mainImage Function

In Shadertoy, your shader is a function called `mainImage`:

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    // your code here
}
```

This function is called once per pixel, every frame. The inputs and outputs:

- `fragCoord` — the pixel coordinates, passed *in* to your function
- `fragColor` — the RGBA color, which you write *out*

The `in` and `out` keywords are explicit about data flow: `fragCoord` is read-only input, `fragColor` is where you write your result. The function returns `void` because the output goes through `fragColor`, not a return value.

1.3.2. Hello World: A Solid Color

The simplest shader: make every pixel red.

The `vec4(1.0, 0.0, 0.0, 1.0)` constructs a 4-component vector: red=1, green=0, blue=0, alpha=1. Every pixel receives the same color, so the screen fills with red.

1.3.3. GLSL Syntax Essentials

GLSL (OpenGL Shading Language) will feel familiar if you've seen C-like syntax, but a few things are worth noting upfront.

Semicolons are required at the end of each statement.

Floats must include a decimal point. Write `1.0`, not `1`. The integer `1` and the float `1.0` are different types, and GLSL is strict about this.

Vector types are built in: `vec2`, `vec3`, `vec4` for 2, 3, and 4 component vectors. Construct them with:

```
vec2 p = vec2(3.0, 4.0);
vec3 color = vec3(1.0, 0.5, 0.0);
vec4 rgba = vec4(1.0, 0.0, 0.0, 1.0);
```

Arithmetic is component-wise. Adding two vectors adds their components:

```
vec2(1.0, 2.0) + vec2(3.0, 4.0)  // = vec2(4.0, 6.0)
```

Scalar-vector operations apply the scalar to each component:

```
2.0 * vec2(1.0, 3.0)  // = vec2(2.0, 6.0)
```

Accessing components uses `.x`, `.y`, `.z`, `.w`:

```
vec2 p = vec2(3.0, 4.0);
float a = p.x;  // 3.0
float b = p.y;  // 4.0
```

For colors, `.r`, `.g`, `.b`, `.a` are synonyms—`color.r` is the same as `color.x`.

Common math functions work as expected: `sin`, `cos`, `abs`, `min`, `max`, `sqrt`, `pow`. These operate on floats, and apply component-wise to vectors:

```
sin(vec2(0.0, 3.14159))  // = vec2(0.0, ~0.0)
```

For loops work as you'd expect:

```
for (int i = 0; i < 5; i++) {
    // body executes with i = 0, 1, 2, 3, 4
}
```

The loop variable is an `int`. Note that some older GPUs require the loop bounds to be constants known at compile time—you can't always loop up to a variable. We'll use loops extensively starting tomorrow.

1.3.4. Uniforms: Global Inputs

Shadertoy provides *uniforms*—global values that are constant across all pixels. Unlike `fragCoord`, which takes a different value at each pixel, a uniform has the same value everywhere. They're how external information (time, screen size, mouse position) gets into your shader.

Uniform	Type	Description
`iResolution`	`vec3`	Viewport size: `(width, height, pixel_aspect_ratio)`
`iTime`	`float`	Seconds since the shader started
`iMouse`	`vec4`	Mouse position and click state

We'll use `iResolution` constantly (for coordinate transforms) and `iTime` for animation.

1.3.5. Animation: Using iTime

Let's make the red channel pulse:

Since `sin(iTime)` oscillates between -1 and 1, the expression `0.5 + 0.5 * sin(iTime)` oscillates between 0 and 1. The screen pulses from black to red.

This is our first *animated* shader—the output depends on time.

1.4. Coordinate Systems

1.4.1. Pixel Coordinates

The input `fragCoord` gives the pixel coordinates of the current pixel. The coordinate system:

- Origin at the **bottom-left** corner
- `fragCoord.x` increases to the right
- `fragCoord.y` increases upward

- Ranges from $(0, 0)$ to (X, Y) where $X \times Y$ is the screen resolution

This is workable, but inconvenient for mathematics. We'd prefer coordinates centered at the origin with a reasonable scale. Let's build up a transformation step by step.

1.4.2. Step 1: Normalize to $[0, 1]^2$

Divide by the resolution to map pixel coordinates to the unit square:

```
vec2 uv = fragCoord / iResolution.xy;
```

Now `uv` ranges from $(0, 0)$ at bottom-left to $(1, 1)$ at top-right.

Since both coordinates are in $[0, 1]$, we can visualize them directly as color:

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    fragColor = vec4(uv.x, uv.y, 0.0, 1.0);
}
```

Black at bottom-left (0,0), red at bottom-right (1,0), green at top-left (0,1), yellow at top-right (1,1).

1.4.3. Step 2: Center the Origin

Subtract $(0.5, 0.5)$ to center the origin:

```
uv = uv - vec2(0.5, 0.5);
```

Now `uv` ranges from $(-0.5, -0.5)$ to $(0.5, 0.5)$, with $(0, 0)$ at the screen center.

1.4.4. Step 3: Aspect Ratio Correction

We've mapped a rectangle of pixels $(X \times Y)$ to the square $[-0.5, 0.5]^2$. This is an affine transformation, not a similarity—it distorts shapes. A circle in our coordinates would render as an ellipse on screen.

To fix this, we scale the x-coordinate by the aspect ratio:

```
uv.x *= iResolution.x / iResolution.y;
```

Now a circle in our coordinates appears as a circle on screen. (When we draw shapes later, try commenting out this line to see the distortion.)

1.4.5. Step 4: Scale to a Useful Range

Finally, scale to a convenient window:

```
vec2 p = uv * 4.0;
```

With a scale factor of 4, our coordinates range roughly from -2 to 2—a good default for visualizing mathematical objects.

1.4.6. The Standard Boilerplate

Putting it together, here's the coordinate setup we'll use throughout the course:

```
vec2 uv = fragCoord / iResolution.xy;   // normalize to [0,1]
uv = uv - vec2(0.5, 0.5);               // center origin
uv.x *= iResolution.x / iResolution.y;  // aspect correction
vec2 p = uv * 4.0;                      // scale
```

From here on, `p` is our mathematical coordinate, centered at the origin, aspect-corrected, with a reasonable range.

1.5. Drawing with Distance

So far we've colored every pixel the same, or colored based on position as a gradient. Now we want to *draw*: to render a shape on screen.

What does it mean to draw a shape? For a simple filled region, we need a rule that tells us, for each pixel: are you inside the shape or not? When inside, we do one thing (say, color yellow). When outside, we do another (color blue). The boundary of the shape is where we switch.

1.5.1. Half-Planes

The simplest shape is a half-plane. Consider the rule: is the y-coordinate greater than 0? This divides the plane into two regions—above and below the x-axis.

```
float L = p.y;

vec3 color;
if (L < 0.0) {
    color = vec3(1.0, 0.0, 0.0);  // red below
} else {
    color = vec3(0.0, 0.0, 1.0);  // blue above
}

fragColor = vec4(color, 1.0);
```

To color left versus right instead, use `p.x` in place of `p.y`.

More generally, a line in the plane has the form $ax + by + c = 0$. This divides the plane into two half-planes: where $ax + by + c < 0$ and where $ax + by + c > 0$.

```
float a = 1.0, b = 1.0, c = 0.0;
float L = a * p.x + b * p.y + c;

vec3 color;
if (L < 0.0) {
    color = vec3(1.0, 0.0, 0.0);  // red
} else {
    color = vec3(0.0, 0.0, 1.0);  // blue
}

fragColor = vec4(color, 1.0);
```

Recall that (a, b) is the normal vector to the line, and c is an offset. Since these are just variables, we can animate them to move the line around:

```
float a = cos(iTime);
float b = sin(iTime);
float c = 0.5 * sin(iTime * 0.7);
```



1.5.2. Circles

Now consider the function $d(p) = |p|$, the distance from the origin. Geometrically, the graph of this function is a cone—zero at the origin, increasing linearly in all directions.

To draw a filled disk of radius r, we could threshold on $d < r$ versus $d \geq r$. But it's cleaner to define $f(p) = |p| - r$. This function is negative inside the circle (where $d < r$) and positive outside (where $d > r$). The circle itself is the level set $f = 0$.

```
float d = length(p);
float r = 1.0;
float f = d - r;

vec3 color;
if (f < 0.0) {
    color = vec3(1.0, 1.0, 0.0);  // yellow inside
} else {
    color = vec3(0.1, 0.1, 0.3);  // dark blue outside
}

fragColor = vec4(color, 1.0);
```



Try commenting out the aspect ratio correction (`uv.x *= ...`) to see the distortion—the circle becomes an ellipse.

To center the circle at a point c instead of the origin, compute distance from c:

```
vec2 center = vec2(1.0, 0.5);
float d = length(p - center);
```

Since `center` and `r` are variables, you can animate them with `iTime` to create moving, pulsing circles.

1.5.3. Drawing a Ring

Our function $f = d - r$ is negative inside the circle and positive outside. To draw a filled disk, we colored based on the sign of f.

But what if we want just the boundary—a ring of some thickness? We want to color one way when f is small in absolute value (near the circle), and a different way when $|f|$ is large (far from the circle).

So we look at $|f| = |d - r|$ and ask: is this less than some threshold ε, or greater? Equivalently, is $|d - r| - \varepsilon$ negative or positive?

```
float d = length(p);
float r = 1.0;
float eps = 0.1;
float f = abs(d - r) - eps;

vec3 color;
if (f < 0.0) {
    color = vec3(1.0, 1.0, 1.0);  // white ring
} else {
    color = vec3(0.1, 0.1, 0.3);  // dark background
}

fragColor = vec4(color, 1.0);
```



1.6. Implicit Curves

We've drawn circles using the distance function $|p| - r$. But circles are just one example of curves defined by an equation. Any equation $F(x, y) = 0$ defines a curve—the set of points satisfying that equation. We can draw it the same way: threshold on $|F|$.

1.6.1. A First Example: The Parabola

Consider $F(x, y) = y - x^2$. The curve $F = 0$ is the parabola $y = x^2$. Points where $F < 0$ lie below the parabola; points where $F > 0$ lie above.

To draw the curve itself, we color pixels where $|F|$ is small:

```
float F = p.y - p.x * p.x;
float eps = 0.1;

vec3 color;
if (abs(F) < eps) {
    color = vec3(1.0, 1.0, 0.0);  // yellow curve
} else {
    color = vec3(0.1, 0.1, 0.3);  // dark background
}

fragColor = vec4(color, 1.0);
```

1.6.2. More Examples

An ellipse: $F(x, y) = \frac{x^2}{a^2} + \frac{y^2}{b^2} - 1$

```
float a = 2.0, b = 1.0;
float F = (p.x*p.x)/(a*a) + (p.y*p.y)/(b*b) - 1.0;
```

A hyperbola: $F(x, y) = \frac{x^2}{a^2} - \frac{y^2}{b^2} - 1$

```
float a = 1.0, b = 1.0;
float F = (p.x*p.x)/(a*a) - (p.y*p.y)/(b*b) - 1.0;
```

The lemniscate of Bernoulli: $(x^2 + y^2)^2 = a^2(x^2 - y^2)$, or $F = (x^2 + y^2)^2 - a^2(x^2 - y^2)$

```
float a = 1.5;
float r2 = dot(p, p);  // x² + y²
float F = r2 * r2 - a * a * (p.x * p.x - p.y * p.y);
```

1.6.3. The Thickness Problem

Look carefully at the parabola. The rendered thickness isn't uniform—it's thinner where the curve is steep, thicker where it's flat. The problem gets worse with more complicated curves, especially those with singularities. Here's the lemniscate:



Notice how the thickness blows up near the origin, where the curve crosses itself.

Why does this happen? The set $|F| < \varepsilon$ contains all points within ε of zero *in the F direction*. But F doesn't measure distance to the curve—it's just some function that happens to be zero on the curve. Where $|\nabla F|$ is large, F changes rapidly, so the band $|F| < \varepsilon$ is narrow. Where $|\nabla F|$ is small, F changes slowly, so the band is wide. At the singular point, $\nabla F = 0$, and the band becomes infinitely wide.

1.6.4. Why Circles Worked

For the circle, we used $f(p) = |p| - r$. This is the *signed distance function*: it measures actual geometric distance to the curve. The gradient of a distance function has magnitude 1 everywhere (it points toward or away from the curve at unit rate). So $|f| < \varepsilon$ really does capture points within distance ε, giving uniform thickness.

This is a fact from differential geometry: $|\nabla d| = 1$ for a distance function d. When we use an arbitrary implicit equation $F = 0$, we lose this property.

1.6.5. Gradient Correction

We can fix the non-uniform thickness by dividing by the gradient magnitude. Instead of thresholding $|F| < \varepsilon$, we threshold

$$\frac{|F|}{|\nabla F|} < \varepsilon.$$

This approximates the signed distance to the curve. The intuition: $|F|/|\nabla F|$ estimates how far you'd need to travel (in the direction F changes fastest) to reach the curve.

For the lemniscate, we compute the gradient analytically:

$$\nabla F = \left(4x(x^2 + y^2) - 2a^2x,\ 4y(x^2 + y^2) + 2a^2y\right)$$

```
float a = 1.5;
float r2 = dot(p, p);
float F = r2 * r2 - a * a * (p.x * p.x - p.y * p.y);

vec2 grad = vec2(
    4.0 * p.x * r2 - 2.0 * a * a * p.x,
    4.0 * p.y * r2 + 2.0 * a * a * p.y
);

float dist = abs(F) / max(length(grad), 0.01);  // avoid division by zero
float eps = 0.05;

vec3 color;
if (dist < eps) {
    color = vec3(1.0, 1.0, 0.0);
} else {
    color = vec3(0.1, 0.1, 0.3);
}

fragColor = vec4(color, 1.0);
```



Compare with the naive version above to see the difference in thickness uniformity.

1.6.6. Animated Curve Families

The lemniscate is part of a one-parameter family called the Cassini ovals, defined by the product of distances from two foci being constant:

$$(x^2 + y^2)^2 - 2c^2(x^2 - y^2) = a^4 - c^4$$

As the parameter a varies relative to the fixed focal distance c, the topology changes: two separate loops when $a < c$, a lemniscate when $a = c$, a single oval when $a > c$.



1.7. Interactivity and Abstraction

So far our shaders respond to time (`iTime`) but not to user input. Shadertoy provides `iMouse` for mouse interaction.

1.7.1. The iMouse Uniform

`iMouse` is a `vec4`:

- `iMouse.xy` — current mouse position (in pixels)
- `iMouse.zw` — position where the mouse was last clicked

For now we'll focus on `iMouse.xy`.

1.7.2. Dragging a Circle

Let's draw a circle centered at the mouse position. Since `iMouse.xy` is in pixel coordinates, we need to normalize it the same way we normalize `fragCoord`:

```
// Normalize fragment coordinate
vec2 uv = fragCoord / iResolution.xy;
uv = uv - vec2(0.5, 0.5);
uv.x *= iResolution.x / iResolution.y;
vec2 p = uv * 4.0;

// Normalize mouse coordinate the same way
vec2 mouse = iMouse.xy / iResolution.xy;
mouse = mouse - vec2(0.5, 0.5);
mouse.x *= iResolution.x / iResolution.y;
mouse = mouse * 4.0;

// Circle centered at mouse
float d = length(p - mouse);
```

```
float r = 0.5;

vec3 color;
if (d < r) {
    color = vec3(1.0, 0.9, 0.2);  // yellow
} else {
    color = vec3(0.1, 0.1, 0.3);
}

fragColor = vec4(color, 1.0);
```



Click and drag to move the circle.

1.7.3. Writing a Helper Function

We just wrote the same four lines of coordinate normalization twice. This is a sign we should write a function.

A GLSL function declares its return type, then the function name, then its parameters with their types:

```
vec2 normalize_coord(vec2 coord) {
    vec2 uv = coord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}
```

Functions must be defined before they're used, so they go above `mainImage`. Here's the overall structure:

```
vec2 normalize_coord(vec2 coord) {
    // normalization logic here
}

void mainImage(out vec4 fragColor, in vec2 fragCoord) {
    vec2 p = normalize_coord(fragCoord);
    vec2 mouse = normalize_coord(iMouse.xy);

    // code using p and mouse
}
```

Now our shader is cleaner, and we won't make mistakes copying the normalization code.

1.7.4. Combining iMouse and iTime: Sun and Earth

Let's make a circle orbit around the mouse position:

```
vec2 p = normalize_coord(fragCoord);
vec2 sun = normalize_coord(iMouse.xy);

// Earth orbits the sun
float orbit_radius = 0.8;
vec2 earth = sun + orbit_radius * vec2(cos(iTime), sin(iTime));

// Draw sun (larger, yellow)
float d_sun = length(p - sun);
// Draw earth (smaller, blue)
float d_earth = length(p - earth);

vec3 color = vec3(0.02, 0.02, 0.05);  // dark background
if (d_sun < 0.3) {
    color = vec3(1.0, 0.9, 0.2);  // yellow sun
}
if (d_earth < 0.15) {
    color = vec3(0.2, 0.5, 1.0);  // blue earth
}

fragColor = vec4(color, 1.0);
```



Drag to move the sun; the earth follows in orbit. (Exercise: add a moon orbiting the earth!)

1.7.5. Mouse as Parameter

The mouse doesn't have to control position—it can control any parameter. A useful pattern: map `iMouse.x` to a parameter range and drag across the screen to explore a family of curves.

The folium of Descartes is the curve $x^3 + y^3 = 3axy$. We can explore its level sets by drawing $x^3 + y^3 - 3axy = c$ for different values of c:

```
vec2 p = normalize_coord(fragCoord);

// Fixed parameter a
float a = 1.5;

// Map mouse x to level set value c in [-2, 2]
float c = mix(-2.0, 2.0, iMouse.x / iResolution.x);

// Folium of Descartes: x³ + y³ - 3axy = c
float F = p.x*p.x*p.x + p.y*p.y*p.y - 3.0*a*p.x*p.y - c;

// Gradient: ∇F = (3x² - 3ay, 3y² - 3ax)
vec2 grad = vec2(3.0*p.x*p.x - 3.0*a*p.y, 3.0*p.y*p.y - 3.0*a*p.x);
float dist = abs(F) / max(length(grad), 0.01);

vec3 color;
if (dist < 0.05) {
    color = vec3(1.0, 1.0, 0.0);
} else {
    color = vec3(0.1, 0.1, 0.3);
}

fragColor = vec4(color, 1.0);
```

Drag left and right to sweep through the level sets and watch the curve topology change.

1.8. Exercises

Homework is organized into four types:

Checkpoints — Short exercises to verify you understood the lecture material. Required for anyone new to shader programming.

Explorations — Open-ended problems that extend the lecture topics. Pick the ones that interest you. If you can do several of these, you're right on track with the course.

Challenges — Problems that may require learning new concepts beyond what was covered in lecture. Attempt these if you skipped the checkpoints and found an exploration or two too easy.

Project — An extended project for someone familiar with shader basics, to make an artwork.

1.8.1. Checkpoints

C1. Solid Colors. Modify the red screen shader to display: (a) green, (b) cyan, (c) a color of your choice using all three RGB channels.

C2. Vertical Split. Modify the half-plane shader to divide the screen into left (red) and right (blue) instead of top and bottom.

C3. Off-Center Circle. Draw a filled circle of radius 0.5 centered at the point $(1, 1)$ instead of the origin.

C4. Pulsing Circle. Make a circle whose radius oscillates between 0.5 and 1.5 over time using `iTime`.

C5. Ring Thickness. Draw a ring (circle outline) centered at the origin. Experiment with different values of `eps` to understand how it controls thickness.

1.8.2. Explorations

E1. Concentric Rings. Draw several concentric rings (circles of different radii, all centered at the origin). Can you color alternate rings differently?

E2. Moon Orbit. Extend the sun-earth shader to add a moon that orbits the earth. The moon should be smaller than the earth and orbit faster.

E3. Your Favorite Curve. Pick an implicit curve from your mathematical experience (or find one online) and render it. Some suggestions: the cardioid $(x^2 + y^2 - ax)^2 = a^2(x^2 + y^2)$, the astroid $x^{2/3} + y^{2/3} = a^{2/3}$, or a rose curve in implicit form. Apply gradient correction for uniform thickness.

E4. Curve Explorer. Take any one-parameter family of curves and build a mouse-controlled explorer (like the folium example). Map `iMouse.x` to the parameter and drag to explore the family.

E5. Two Circles. Draw two filled circles at different positions. What happens when they overlap? Can you make one "in front of" the other? Can you make the intersection a different color, like a Venn diagram?

1.8.3. Challenges

H1. Parabola Graphing Calculator. Build an interactive graphing calculator for the parabola $y = ax^2 + bx + c$. Requirements: - Draw coordinate axes (the lines $x = 0$ and $y = 0$) - Draw the parabola using implicit curve techniques - Find the roots (where $y = 0$) and draw small circles around them - Use mouse position to control two of the coefficients (e.g., a and b, with c fixed, or b and c with a fixed)

As you drag the mouse, the parabola should reshape and the root indicators should move (or appear/disappear as roots become real or complex).

H2. Elliptic Curve Explorer. Elliptic curves in Weierstrass form are $y^2 = x^3 + ax + b$. Build a shader where the mouse position controls (a, b). Use gradient correction for uniform thickness. The *discriminant* $\Delta = 4a^3 + 27b^2$ determines whether the curve is smooth ($\Delta \neq 0$) or singular ($\Delta = 0$). Can you display the current value of Δ somehow, or change the curve's color when it becomes singular?

H3. Signed Distance Functions. For a filled circle, $f(p) = |p| - r$ is the *signed* distance function: negative inside, positive outside, with $|f|$ giving the actual distance to the boundary. What is the signed distance function for a half-plane? For an axis-aligned rectangle? Implement both and draw them with uniform-thickness boundaries. Note: when you have the true signed distance function, you don't need the gradient correction trick—that's the payoff for computing the right thing from the start!

H4. Smooth Blending. When two circles overlap, we currently just draw one on top of the other. Research *smooth minimum* functions (e.g., `smin`) that blend distance fields smoothly. Draw two circles that "melt together" where they meet.

H5. Inversion. Circle inversion is the map $p \mapsto p/|p|^2$. Apply this transformation to your coordinate p before drawing a shape. What happens to a line? What happens to a circle not passing through the origin? Experiment with different shapes.

1.8.4. Project: Grid Patterns

This extended project introduces a powerful technique—using modular arithmetic to repeat patterns across the plane. We'll build up the machinery carefully, since we'll use it again in Day 2 to create grids of Julia sets.

1.8.4.1. Part 1: Setting Up a Grid of Square Cells

We want to tile the screen with square cells—say, 4 cells across. The challenge: the screen isn't square, so we need to handle the aspect ratio.

Let's say we want `N` columns of cells. Each cell has width $L = \text{screen\_width}/N$ in pixels, and since cells are square, height L as well. The number of rows depends on the screen's aspect ratio.

Working in our normalized coordinates (after aspect correction), the screen spans roughly $[-2 \cdot \text{aspect}, 2 \cdot \text{aspect}]$ in x and $[-2, 2]$ in y. If we want cells of side length L in these coordinates:

```
float aspect = iResolution.x / iResolution.y;
float N = 5.0;  // number of columns
float L = (4.0 * aspect) / N;  // cell size in our coordinate system
```

Now each cell is an $L \times L$ square.

1.8.4.2. Part 2: Cell Coordinates and Identity

For each pixel, we want two things:

1. **Which cell are we in?** Integer coordinates (i, j) identifying the cell.
2. **Where in the cell are we?** Local coordinates ranging from $-L/2$ to $L/2$, with $(0, 0)$ at the cell center.

```
vec2 cell_id = floor(p / L);
vec2 cell_p = mod(p + vec2(L/2.0, L/2.0), L) - vec2(L/2.0, L/2.0);
```

The `cell_id` tells us which cell; the `cell_p` gives local coordinates within that cell.

If we want local coordinates normalized to $[-1, 1]$ (useful for drawing things at a standard scale), we can rescale:

```
vec2 local = cell_p / (L / 2.0);  // now in [-1, 1] x [-1, 1]
```

This is exactly the setup we'll need for Day 2, where each cell will contain a Julia set with its own coordinate system.

1.8.4.3. Part 3: Drawing in Each Cell

Now draw something using the local coordinates. A filled circle at the center of each cell:

```
float d = length(cell_p);
float r = L * 0.4;  // radius relative to cell size

vec3 color;
if (d < r) {
    color = vec3(1.0, 1.0, 0.0);
} else {
    color = vec3(0.1, 0.1, 0.3);
}
```

Try changing `N` to get more or fewer columns. The cells stay square regardless of screen shape.

1.8.4.4. Part 4: Varying by Cell

The `cell_id` lets each cell behave differently. Some ideas:

Checkerboard background:

```
float checker = mod(cell_id.x + cell_id.y, 2.0);
vec3 bg = mix(vec3(0.2, 0.2, 0.3), vec3(0.3, 0.2, 0.2), checker);
```

Radius varying by cell:

```
float r = L * (0.2 + 0.15 * mod(cell_id.x + cell_id.y, 3.0));
```

Wave animation:

```
float cell_dist = length(cell_id);
float r = L * (0.3 + 0.1 * sin(iTime * 2.0 - cell_dist * 0.5));
```

1.8.4.5. Part 5: Design Challenge

Design a grid-based pattern that you find visually interesting. Some directions:

Connecting shapes: Draw shapes that connect across cell boundaries. Quarter-circles in each corner create a continuous network. What implicit curves tile seamlessly?

Alternating motifs: Use `cell_id` to alternate between different shapes—circles in some cells, rings in others, or different orientations.

Color fields: Map `cell_id` to colors using distance from origin, stripes, or a palette.

Phase shifts: Animate cells with different phase offsets to create waves or ripples.

Using local coordinates: Draw something more complex in each cell using the $[-1, 1]$ local coordinate system—perhaps a small implicit curve, or a pattern that changes based on `cell_id`.

The goal is to produce an image you'd be happy to hang on a wall.

1.8.5. Project: Fourier Epicycles

This project builds a visualization of Fourier series using epicycles—circles whose centers sit on the circumferences of other circles. This is how Ptolemy modeled planetary motion, and it turns out to be exactly how Fourier series work geometrically.

1.8.5.1. Part 1: The Idea

Any periodic function can be written as a sum of sines and cosines. Geometrically, $\sin(n\omega t)$ and $\cos(n\omega t)$ describe a point moving around a circle of frequency $n\omega$. Adding these components corresponds to stacking circles: each circle's center rides on the previous circle's edge.

For example, the square wave has Fourier series:

$$f(t) = \sum_{n=1,3,5,\ldots} \frac{1}{n} \sin(n\omega t)$$

This means circles with: - Radii: $1, \frac{1}{3}, \frac{1}{5}, \frac{1}{7}, \ldots$ - Frequencies: $\omega, 3\omega, 5\omega, 7\omega, \ldots$

The more terms we add, the closer the final point's y-coordinate approximates a square wave.

1.8.5.2. Part 2: Drawing Circles

Start by drawing a chain of circles. Each circle is centered at the current position, and the next position is computed by moving along the circle:

```
vec2 pos = vec2(0.0, 0.0);  // start at origin

for (int i = 0; i < N; i++) {
    int n = 2 * i + 1;  // 1, 3, 5, 7, ...
    float r = scale / float(n);
    float freq = float(n) * omega;

    // Draw circle at current position
    float d_circle = abs(length(p - pos) - r);
    if (d_circle < 0.02) {
        // color the circle
    }

    // Move to next position
    pos = pos + r * vec2(cos(freq * iTime), sin(freq * iTime));
}

// Draw final point
float d_point = length(p - pos);
if (d_point < 0.08) {
    // bright color
}
```

Try this with `N = 1`, then `N = 3`, then `N = 7`. Watch how more circles create more complex motion.

1.8.5.3. Part 3: The Line Segment SDF

To draw the arms connecting circle centers, we need the signed distance function for a line segment. Given endpoints a and b, the distance from point p to the segment is:

```
float sd_segment(vec2 p, vec2 a, vec2 b) {
    vec2 pa = p - a;
    vec2 ba = b - a;
    float t = clamp(dot(pa, ba) / dot(ba, ba), 0.0, 1.0);
    return length(pa - ba * t);
}
```

The math: we project $p - a$ onto the line direction $b - a$, clamp to $[0, 1]$ to stay within the segment, then measure the distance to that closest point.

1.8.5.4. Part 4: Connecting the Arms

Now modify your loop to also draw line segments:

```
vec2 pos = vec2(0.0, 0.0);

for (int i = 0; i < N; i++) {
    int n = 2 * i + 1;
    float r = scale / float(n);
    float freq = float(n) * omega;

    vec2 next_pos = pos + r * vec2(cos(freq * iTime), sin(freq * iTime));

    // Draw circle
    float d_circle = abs(length(p - pos) - r);
    if (d_circle < 0.02) {
        // faint circle color
    }

    // Draw arm from pos to next_pos
    float d_arm = sd_segment(p, pos, next_pos);
    if (d_arm < 0.015) {
        // arm color
    }

    pos = next_pos;
}
```

1.8.5.5. Part 5: Polish and Explore

Now make it beautiful:

Fading circles: Later circles are smaller and less important. Fade their brightness:

```
float fade = 1.0 - float(i) / float(N);
```

Color variation: Color circles differently based on their index, or based on their frequency.

Different waves: The square wave uses odd harmonics with $1/n$ coefficients. Try: - Triangle wave: odd harmonics with $1/n^2$ coefficients (alternating signs) - Sawtooth wave: all harmonics with $1/n$ coefficients

Mouse control: Map `iMouse.x` to the number of terms, so dragging adds or removes circles.

The goal: create a mesmerizing animation that reveals the geometry hidden inside Fourier series.

2. Day 2: Dynamics

2.1. Overview

Today we'll learn how to do per-pixel computation in shaders, to quickly produce images of compelx mathematical objects. In the homework today, you'll create your own version of the classic mandelbrot fractal zoom:



Today we explore two kinds of iterative systems:

1. **Complex dynamics**: Iterating holomorphic maps gives us the Mandelbrot set, Julia sets, and their cousins
2. **Circle inversion**: Iterating geometric transformations gives us the Apollonian gasket

Both share the same GPU-friendly structure: each pixel asks "what happens when I iterate from here?" No pixel depends on any other—perfect for parallel computation.

Along the way, we'll learn to implement complex arithmetic in GLSL and organize geometric data using structs.

2.2. Complex Numbers in GLSL

The complex numbers $\mathbb{C}$ are the plane equipped with a multiplication operation. Today we implement that algebra in GLSL.

2.2.1. Representation

A complex number $z = a + bi$ is naturally represented as a 2D vector:

```
vec2 z = vec2(a, b);  // represents a + bi
```

We'll consistently use the convention that `z.x` is the real part and `z.y` is the imaginary part.

2.2.2. Arithmetic

Addition of complex numbers is componentwise—exactly what GLSL's built-in `+` does for vectors. No helper function needed.

Multiplication is more interesting. In GLSL, the `*` operator on vectors is componentwise: `vec2(a,b) * vec2(c,d)` gives `vec2(a*c, b*d)`. This is *not* complex multiplication! We need to implement the correct formula ourselves.

Recall $(a + bi)(c + di) = (ac - bd) + (ad + bc)i$:

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(
        z.x * w.x - z.y * w.y,
        z.x * w.y + z.y * w.x
    );
}
```

This is the FOIL pattern with $i^2 = -1$ giving the minus sign in the real part.

2.2.3. Magnitude

The magnitude of $z = a + bi$ is the distance from the origin:

$$|z| = \sqrt{a^2 + b^2}$$

We can implement this directly:

```
float cabs(vec2 z) {
    return sqrt(z.x * z.x + z.y * z.y);
}
```

But consider: we'll often have conditions like "is $|z|$ bigger than 2?" rather than needing the actual magnitude. In these cases, we can check $|z|^2 > 4$ instead of $|z| > 2$—same answer, but no square root. When you're doing this check millions of times per frame (once per pixel, 60 frames per second), avoiding unnecessary square roots adds up.

So we define the squared magnitude:

```
float cabs2(vec2 z) {
    return z.x * z.x + z.y * z.y;
}
```

If we want to be even more efficient, we can use GLSL's built-in dot product, which computes exactly this sum of products:

```
float cabs2(vec2 z) {
    return dot(z, z);  // a² + b²
}
```

2.3. The Mandelbrot Set

The Mandelbrot set is perhaps the most iconic fractal—its shape is instantly recognizable, and its discovery in 1980 helped launch the era of computer-generated mathematical visualization. The definition is remarkably simple.

2.3.1. Definition

Fix a complex number c. Starting from $z_0 = 0$, define a sequence by iterating:

$$z_{n+1} = z_n^2 + c$$

For some values of c, this sequence stays bounded forever. For others, it escapes to infinity. The **Mandelbrot set** $\mathcal{M}$ is the set of all c for which the sequence remains bounded.

That's it! One quadratic formula, iterated. The intricate structure of the Mandelbrot set emerges entirely from this simple rule.

2.3.2. Rendering Strategy

To draw the Mandelbrot set, we test each pixel: is this value of c in $\mathcal{M}$ or not?

This is exactly the kind of question shaders excel at. Each pixel performs its own independent calculation—no pixel needs information from any other pixel. The entire image can be computed in parallel, with thousands of GPU cores each testing their own value of c simultaneously.

But there's a problem: the definition involves iterating "forever" and checking if the sequence "stays bounded." We can't iterate infinitely, and we can't wait forever to decide. We need a practical criterion for when to stop.

2.3.3. The Escape Radius

We need two facts that make efficient rendering possible.

Fact 1. The Mandelbrot set is contained in the disk of radius 2. That is, if $|c| > 2$, then $c \notin \mathcal{M}$.

Proof. [TODO] □

Fact 2. If $|c| \leq 2$ and $|z_n| > 2$ for some n, then the orbit escapes to infinity (so $c \notin \mathcal{M}$).

Proof. [TODO] □

This gives us a stopping criterion: once $|z_n| > 2$, we know c is not in the Mandelbrot set. We don't need to keep iterating.

Together, these facts justify the **escape-time algorithm**: iterate until either $|z_n| > 2$ (escaped, not in $\mathcal{M}$) or we hit a maximum iteration count (probably in $\mathcal{M}$).

2.3.4. The Escape-Time Algorithm

These two facts give us our algorithm:

1. For each pixel, let c be the corresponding complex number
2. Start with $z = 0$
3. Iterate $z \mapsto z^2 + c$
4. If $|z| > 2$, stop—this point escapes (not in $\mathcal{M}$)
5. If we reach a maximum iteration count without escaping, assume bounded (in $\mathcal{M}$)

The core of this is a loop:

```
vec2 z = vec2(0.0, 0.0);
int max_iter = 100;
int iter;

for (iter = 0; iter < max_iter; iter++) {
    if (cabs2(z) > 4.0) break;  // |z|² > 4 means |z| > 2
    z = cmul(z, z) + c;
}
```

After this loop, `iter` tells us what happened: if `iter == max_iter`, we never escaped (probably in $\mathcal{M}$). Otherwise, we escaped on iteration `iter`.

2.3.5. Binary Coloring

The simplest approach: color points black if they're in the set, white if they escaped.

```
vec3 color;
if (iter == max_iter) {
    color = vec3(0.0);  // In the set: black
} else {
    color = vec3(1.0);  // Escaped: white
}
```

2.3.6. Full Implementation

Putting it together with our coordinate setup:

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    // Coordinate setup
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;

    // Scale and center to show the Mandelbrot set
    // By Fact 1, the set lies in |c| ≤ 2
    vec2 c = uv * 4.0;
    c.x -= 0.5;  // shift left to center the interesting part

    // Mandelbrot iteration
    vec2 z = vec2(0.0, 0.0);
    int max_iter = 100;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (cabs2(z) > 4.0) break;
        z = cmul(z, z) + c;
    }

    // Binary coloring
    vec3 color;
    if (iter == max_iter) {
        color = vec3(0.0);  // In the set: black
    } else {
        color = vec3(1.0);  // Escaped: white
    }

    fragColor = vec4(color, 1.0);
}
```



There it is—the Mandelbrot set in black and white!

2.3.7. Coloring by Iteration Count

Black and white shows the set, but we're throwing away information. The number of iterations before escape tells us how "close" a point is to the boundary—points that escape after 5 iterations are different from points that escape after 50.

Let's use `iter` to create a gradient:

```
vec3 color;
if (iter == max_iter) {
    color = vec3(0.0);
} else {
    float t = float(iter) / float(max_iter);
    color = vec3(t);
}
```



Now we see structure! The boundary reveals intricate detail—tendrils, spirals, bulbs. Points near the boundary take many iterations to escape (bright), while points far away escape quickly (dark).

2.3.8. Color Palettes

Grayscale works, but we can do better. A common technique uses cosines to create smooth, cycling color palettes:

```
vec3 palette(float t) {
    vec3 a = vec3(0.5, 0.5, 0.5);
    vec3 b = vec3(0.5, 0.5, 0.5);
    vec3 c = vec3(1.0, 1.0, 1.0);
    vec3 d = vec3(0.00, 0.33, 0.67);
    return a + b * cos(6.28318 * (c * t + d));
}
```

The parameters `a`, `b`, `c`, `d` control the palette's character. The vector `d` shifts the phase of each color channel, creating different hues. Try `d = vec3(0.00, 0.10, 0.20)` for blues and purples, or `d = vec3(0.30, 0.20, 0.20)` for warmer tones.



The color bands correspond to iteration counts—regions of the same color escaped after the same number of iterations. You'll notice the bands have sharp edges. In the exercises, we'll show you a technique called *smooth coloring* that interpolates between iteration counts, eliminating the banding for even smoother gradients.

2.4. Other Escape-Time Fractals

The Mandelbrot set is one example of an escape-time fractal, but the same algorithm works for many other iterated systems. We just swap out the iteration formula (and sometimes the escape condition). Let's explore a few.

2.4.1. Julia Sets

The Mandelbrot set asks: for which values of c does the orbit of 0 stay bounded? We can ask a different question: for a *fixed* c, which *starting points* z_0 have bounded orbits?

Fix a complex number c. The **filled Julia set** K_c is the set of all starting points z_0 for which the iteration

$$z_{n+1} = z_n^2 + c$$

remains bounded.

Same iteration, different question. For the Mandelbrot set, we vary c and always start at $z_0 = 0$. For a Julia set, we fix c and vary z_0.

The code change is minimal:

```
// Mandelbrot: c varies, z starts at 0
vec2 c = p;
vec2 z = vec2(0.0, 0.0);

// Julia: c is fixed, z starts at pixel position
vec2 c = vec2(-0.7, 0.27015);  // fixed parameter
vec2 z = p;
```



Different values of c produce dramatically different Julia sets. In the exercises, you'll build an interactive explorer that lets you click anywhere on the Mandelbrot set to see the corresponding Julia set:



2.4.2. Others

The Mandelbrot set uses $z^2 + c$. What about $z^3 + c$? Or $z^4 + c$?

For z^3, we can compose our existing `cmul` function:

```
vec2 ccube(vec2 z) {
    return cmul(cmul(z, z), z);  // z · z · z
}
```

Higher powers give higher-order rotational symmetry: $z^3 + c$ has 3-fold symmetry, $z^4 + c$ has 4-fold symmetry, and so on.

For some more variety, the **Burning Ship fractal** modifies the Mandelbrot iteration by taking the absolute value of the imaginary part after each squaring:

$$z_{n+1} = (\text{Re}(z_n^2) + i|\text{Im}(z_n^2)|) + c$$

2.4.3. The Pattern

All these fractals share the same structure:

1. **Iterate** some function $f(z, c)$
2. **Check escape**: has $|z|$ exceeded some threshold?
3. **Color** based on iteration count

Changing the iteration function changes the fractal. The exercises include more variations for you to try.

2.5. Circle Inversion

We shift gears from complex dynamics to geometric transformations. Circle inversion is a classical operation that turns circles into circles (or lines), preserves angles, and creates beautiful fractal patterns when iterated.

2.5.1. Definition

Inversion in the unit circle maps a point $\mathbf{p}$ to:

$$\text{inv}(\mathbf{p}) = \frac{\mathbf{p}}{|\mathbf{p}|^2}$$

What does this do geometrically? The inverted point lies on the same ray from the origin as $\mathbf{p}$, but at distance $1/r$ instead of r.

- Points inside the unit circle map to points outside
- Points outside map to points inside

- Points on the unit circle stay fixed
- The origin maps to "infinity"

2.5.2. Implementation

```
vec2 invert(vec2 p) {
    return p / dot(p, p);
}
```

2.5.3. Visualizing Inversion

To see what inversion does, let's draw some shapes and their images. We'll draw the unit circle (gray), plus a vertical line and a circle (yellow). To make it clearer, we'll toggle between showing the original shapes and their inversions:

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    // Compute the inversion of p
    vec2 p_inv = invert(p);

    // Toggle between original and inverted every second
    float time = fract(iTime * 0.5);
    vec2 q;
```

```
    if (time < 0.5) {
        q = p;       // original
    } else {
        q = p_inv;   // inverted
    }

    vec3 color = vec3(0.1, 0.1, 0.15);

    // Draw the unit circle (the inversion circle)
    float d_unit = abs(length(p) - 1.0);
    if (d_unit < 0.02) color = vec3(0.5, 0.5, 0.5);

    // Draw a vertical line at x = 2
    if (abs(q.x - 2.0) < 0.02) color = vec3(1.0, 1.0, 0.0);

    // Draw a horizontal line at y = 1.5
    if (abs(q.y - 1.5) < 0.02) color = vec3(1.0, 1.0, 0.0);

    // Draw a circle centered at (2, 0) with radius 0.5
    float d_circle = abs(length(q - vec2(2.0, 0.0)) - 0.5);
    if (d_circle < 0.02) color = vec3(1.0, 1.0, 0.0);

    fragColor = vec4(color, 1.0);
}
```



Watch the lines become circles! A line not passing through the origin inverts to a circle that *does* pass through the origin. The circle inverts to another circle (with a different center and radius).

GLSL Shortcuts: mix and step

The toggle logic can be written more compactly using built-in functions:

- `step(edge, x)` returns 0 if `x < edge`, otherwise 1
- `mix(a, b, t)` linearly interpolates: returns `a` when `t = 0`, `b` when `t = 1`

```
float t = step(0.5, fract(iTime * 0.5));
vec2 q = mix(p, p_inv, t);
```

This is a common pattern for toggling or smoothly transitioning between states.

2.5.4. Key Properties

Circle inversion maps circles to circles (or to lines, if the circle passes through the center). It's *conformal*—it preserves angles between curves. And it's *involutive*: applying inversion twice returns to the original point.

2.5.5. Inverting a Grid

For a more dramatic visualization, let's invert a whole grid of lines. The function `mod(q, 0.5)` gives the position of `q` within a repeating 0.5×0.5 cell—when either component is near zero, we're on a grid line:

```
vec2 grid = mod(q, 0.5);
if (grid.x < 0.02 || grid.y < 0.02) color = vec3(1.0, 1.0, 0.0);
```

(If you did the grid project from Day 1, this is familiar!)

Using the same toggle shader structure as before, just replacing the individual shapes with this grid:



2.6. Structs

So far our `invert` function only works for the unit circle at the origin. What if we want to invert through a different circle?

2.6.1. General Circle Inversion

For a circle with center $\mathbf{c}$ and radius R, inversion maps a point $\mathbf{p}$ to:

$$\text{inv}(\mathbf{p}) = \mathbf{c} + R^2 \frac{\mathbf{p} - \mathbf{c}}{|\mathbf{p} - \mathbf{c}|^2}$$

The idea is the same as before: the inverted point lies on the ray from $\mathbf{c}$ through $\mathbf{p}$, at distance R^2/r from the center (where $r = |\mathbf{p} - \mathbf{c}|$). When $\mathbf{c} = \mathbf{0}$ and $R = 1$, this reduces to our earlier formula $\mathbf{p}/|\mathbf{p}|^2$.

In code:

```
vec2 invertInCircle(vec2 p, vec2 center, float radius) {
    vec2 d = p - center;
    return center + radius * radius * d / dot(d, d);
}
```

This works, but notice we need to pass *two* things (`center` and `radius`) to describe *one* object (a circle). If we're working with multiple circles, every function call needs `center1, radius1, center2, radius2, ...`—it gets verbose and error-prone.

2.6.2. Defining a Struct

GLSL lets us bundle related data into a **struct**:

```
struct Circle {
    vec2 center;
    float radius;
};
```

Now `Circle` is a type, just like `vec2` or `float`. We can create circles and access their fields:

```
Circle c;
c.center = vec2(1.0, 0.5);
c.radius = 0.7;

// Or initialize directly:
Circle c = Circle(vec2(1.0, 0.5), 0.7);
```

2.6.3. Inversion with Structs

Now our inversion function takes a `Circle`:

```
vec2 invert(vec2 p, Circle c) {
    vec2 d = p - c.center;
    return c.center + c.radius * c.radius * d / dot(d, d);
}
```

Much cleaner! And when we're working with multiple circles, we can pass them around as single objects.

2.6.4. Demo: Moving Circle

Let's animate a circle moving around and watch how the inversion changes. The key new elements are animating the circle's center and radius:

```
// Animate the inversion circle
Circle inv_circle;
inv_circle.center = vec2(sin(iTime) * 0.5, cos(iTime * 0.7) * 0.5);
inv_circle.radius = 1.0 + 0.3 * sin(iTime * 1.3);

// Compute inversion
vec2 p_inv = invert(p, inv_circle);
```



As the circle moves and breathes, the inverted grid warps and flows.

2.7. The Apollonian Gasket

The Apollonian gasket is a classical fractal arising from circle packing. It's named after Apollonius of Perga (~200 BCE), who studied the problem of finding circles tangent to three given circles.

2.7.1. The Setup

Start with four mutually tangent circles: three "inner" circles that touch each other pairwise, all enclosed by one "outer" circle that touches all three.

Let's define these circles using our `Circle` struct. We'll place three circles of radius r with centers forming an equilateral triangle, all tangent to each other and to an outer circle:

```
// Three inner circles, mutually tangent, plus outer circle
// For circles of radius r to be mutually tangent, their centers
// must be 2r apart. This forms an equilateral triangle with side 2r.
float r = 1.0;
float triSide = 2.0 * r;  // distance between inner circle centers
float circumradius = triSide / sqrt(3.0);  // distance from origin to centers

Circle c1 = Circle(vec2(0.0, circumradius), r);
Circle c2 = Circle(vec2(-circumradius * 0.866, -circumradius * 0.5), r);
Circle c3 = Circle(vec2(circumradius * 0.866, -circumradius * 0.5), r);
Circle outer = Circle(vec2(0.0, 0.0), circumradius + r);
```

To draw these circles, we need a distance function:

```
float distToCircle(vec2 p, Circle c) {
    return abs(length(p - c.center) - c.radius);
}
```

This returns the distance from `p` to the circle's boundary—zero on the circle, positive elsewhere.

Let's draw our starting configuration:

```
vec3 color = vec3(0.1, 0.1, 0.15);

// Draw all four circles
if (distToCircle(p, c1) < 0.03) color = vec3(1.0, 0.3, 0.3);
if (distToCircle(p, c2) < 0.03) color = vec3(0.3, 1.0, 0.3);
if (distToCircle(p, c3) < 0.03) color = vec3(0.3, 0.3, 1.0);
if (distToCircle(p, outer) < 0.03) color = vec3(1.0, 1.0, 1.0);
```

Three colored circles inside a white outer circle, all mutually tangent.

2.7.2. From Drawing to Iteration

The gaps between circles are curvilinear triangles. The Apollonian gasket fills each gap with a circle tangent to its three neighbors, then fills the new gaps, and so on forever.

Here's the key insight: we can generate this structure by *iterating inversions*. If a point is inside one of the inner circles, invert through that circle—this "pushes" it out. If a point is outside the outer circle, invert through the outer circle—this "pulls" it in.

We need to check: - Is **p** inside circle `c1`, `c2`, or `c3`? (distance from center < radius) - Is **p** outside circle `outer`? (distance from center > radius)

```
bool isInside(vec2 p, Circle c) {
    return length(p - c.center) < c.radius;
}

bool isOutside(vec2 p, Circle c) {
    return length(p - c.center) > c.radius;
}
```

The iteration: keep inverting until the point lands in a "gap" (inside outer, outside all inner circles) or we hit a maximum iteration count.

2.7.3. Full Implementation

```
struct Circle {
    vec2 center;
    float radius;
};
```

```
vec2 invert(vec2 p, Circle c) {
    vec2 d = p - c.center;
    return c.center + c.radius * c.radius * d / dot(d, d);
}

float distToCircle(vec2 p, Circle c) {
    return abs(length(p - c.center) - c.radius);
}

bool isInside(vec2 p, Circle c) {
    return length(p - c.center) < c.radius;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 6.0;

    // Setup circles with correct geometry
    float r = 1.0;
    float triSide = 2.0 * r;
    float circumradius = triSide / sqrt(3.0);

    Circle c1 = Circle(vec2(0.0, circumradius), r);
    Circle c2 = Circle(vec2(-circumradius * 0.866, -circumradius * 0.5), r);
    Circle c3 = Circle(vec2(circumradius * 0.866, -circumradius * 0.5), r);
    Circle outer = Circle(vec2(0.0, 0.0), circumradius + r);

    // Iterate inversions
    int max_iter = 50;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (isInside(p, c1)) {
            p = invert(p, c1);
        } else if (isInside(p, c2)) {
            p = invert(p, c2);
        } else if (isInside(p, c3)) {
            p = invert(p, c3);
        } else if (!isInside(p, outer)) {
            p = invert(p, outer);
        } else {
            break;  // In the gap—done!
        }
    }

    // Color by iteration count
    float t = float(iter) / float(max_iter);
```

```
    vec3 color = palette(t);

    // Draw circle boundaries
    float dMin = min(min(distToCircle(p, c1), distToCircle(p, c2)),
                     min(distToCircle(p, c3), distToCircle(p, outer)));
    if (dMin < 0.02) color = vec3(1.0);

    fragColor = vec4(color, 1.0);
}
```

Missing Demo

Shader demo `day2/apollonian-iterated` not found.

2.7.4. Visualizing the Limit Set

The **limit set** of the Apollonian gasket is the fractal boundary—the set of points that never escape to the fundamental domain, no matter how many iterations. Points near the limit set take many iterations before landing in a gap.

We can emphasize the limit set by adjusting our coloring. Instead of using a color palette, we use a nonlinear function that suppresses low iteration counts (the "background") and brightens high iteration counts (near the fractal):

```
float t = float(iter) / float(max_iter);
vec3 color = 30.0 * vec3(pow(t, 2.0));
```

The squaring suppresses points that escape quickly, while the factor of 30 boosts the brightness of points near the limit set.



2.8. Exercises

2.8.1. Checkpoints

C1. Julia Set. Modify the Mandelbrot shader to render a Julia set. Fix `c = vec2(-0.7, 0.27015)` and initialize `z` from the pixel position instead of zero. Verify you get an intricate, connected fractal.

C2. Cubic Mandelbrot. Change the iteration from $z^2 + c$ to $z^3 + c$. You'll need to implement complex cubing:

```
vec2 ccube(vec2 z) {
    return cmul(cmul(z, z), z);
}
```

What symmetry do you observe?

C3. Apollonian Animation. Animate the Apollonian gasket by letting the maximum iteration count grow with time. Use `int max_iter = int(mod(iTime * 5.0, 50.0)) + 1;` so the fractal "builds up" from the four starting circles to the full gasket, then resets. Watch how each iteration reveals a new layer of circles in the gaps.

C4. Colorize a Fractal. Take any of the black-and-white fractals from today (Mandelbrot, Julia, Burning Ship, or Apollonian) and add color using the cosine palette:

```
vec3 palette(float t) {
    vec3 a = vec3(0.5, 0.5, 0.5);
    vec3 b = vec3(0.5, 0.5, 0.5);
    vec3 c = vec3(1.0, 1.0, 1.0);
    vec3 d = vec3(0.00, 0.33, 0.67);
    return a + b * cos(6.28318 * (c * t + d));
}
```

Experiment with different values of `d` to shift the hues.

C5. Circle Art. Create an image with several circles of different sizes scattered across the screen. Color a pixel based on whether it's inside zero, one, two, or more circles. (Hint: use `isInside` and count how many circles contain each point.)

2.8.2. Explorations

E1. Julia Explorer (Mouse). Make the Julia parameter `c` follow the mouse position. Map `iMouse.xy` to a reasonable region of the complex plane (say, $[-2, 2] \times [-2, 2]$). Drag around and watch the Julia set morph!

E2. Julia Animation. Animate the parameter `c` along a path in the complex plane. Try a circle:

```
float angle = iTime * 0.3;
vec2 c = 0.7885 * vec2(cos(angle), sin(angle));
```

Or trace the boundary of the main cardioid of the Mandelbrot set—every point on this curve gives a Julia set with a parabolic fixed point:

```
vec2 cardioid(float t) {
    vec2 eit = vec2(cos(t), sin(t));
    vec2 z = (vec2(2.0, 0.0) - eit) / 4.0;
    return cmul(eit, z);
}

vec2 c = cardioid(iTime * 0.5);
```

Watch the Julia set continuously transform. What happens when `c` crosses from inside to outside the Mandelbrot set?

E3. Other Escape-Time Fractals. Implement one or more of these variations on the Mandelbrot iteration:

- **Burning Ship**: $z \leftarrow z^2$, then $\mathrm{Im}(z) \leftarrow |\mathrm{Im}(z)|$, then $z \leftarrow z + c$
- **Tricorn** (Mandelbar): $z_{n+1} = \bar{z}_n^2 + c$ where $\bar{z}$ is the complex conjugate
- **Celtic**: $z_{n+1} = |\mathrm{Re}(z_n^2)| + i\,\mathrm{Im}(z_n^2) + c$

For each, figure out how to translate the mathematical formula into GLSL. The escape condition ($|z| > 2$) stays the same.

E4. Smooth Coloring. The iteration count is an integer, so coloring by iteration gives discrete bands. But when a point escapes, it doesn't land exactly on the escape radius—it overshoots. We can use *how much* it overshot to interpolate between iteration counts.

The idea: if $|z_n| > 2$, the "true" fractional iteration where $|z| = 2$ is approximately:

$$n_{\text{smooth}} = n - \frac{\log(\log|z_n| / \log 2)}{\log 2}$$

This comes from the fact that near escape, $|z_{n+1}| \approx |z_n|^2$, so $\log|z|$ roughly doubles each iteration.

Implement this for the Mandelbrot set:

```
if (iter < max_iter) {
    float log_zn = log(cabs2(z)) / 2.0;  // log|z|
    float nu = log(log_zn / log(2.0)) / log(2.0);
    float smooth_iter = float(iter) + 1.0 - nu;
    t = smooth_iter / float(max_iter);
}
```

The bands should disappear, replaced by smooth gradients.

E5. Apollonian Coloring. Modify the Apollonian gasket to color by *which circle* was last inverted through, instead of iteration count. Use a different color for each of the four circles. What patterns emerge?

E6. Apollonian Variations. The Apollonian gasket works with *any* four mutually tangent circles—the symmetric configuration we used is just one example. Descartes' Circle Theorem tells

us: if four circles are mutually tangent with curvatures k_1, k_2, k_3, k_4 (where curvature = 1/radius, negative for the outer circle), then:

$$(k_1 + k_2 + k_3 + k_4)^2 = 2(k_1^2 + k_2^2 + k_3^2 + k_4^2)$$

Experiment with different configurations: - Change the radii of the three inner circles (they don't have to be equal!) - Use Descartes' theorem to find an outer circle tangent to three given inner circles - What happens if the circles overlap instead of being tangent?

2.8.3. Challenges

H1. Julia Explorer (Full). Build an interactive tool: display the Mandelbrot set, and wherever the user clicks, show the Julia set for that parameter overlaid or side-by-side. This requires: - Rendering Mandelbrot in one region - Reading click position from `iMouse` - Rendering Julia for that `c` in another region (or blended on top)

H2. Newton Fractal. The Newton fractal comes from applying Newton's method to find roots of a polynomial. For $f(z) = z^3 - 1$, Newton's iteration is:

$$z_{n+1} = z_n - \frac{f(z_n)}{f'(z_n)} = z_n - \frac{z_n^3 - 1}{3z_n^2} = \frac{2z_n^3 + 1}{3z_n^2}$$

Iterate this and color based on *which root* the orbit converges to (the three cube roots of unity: 1, $e^{2\pi i/3}$, $e^{4\pi i/3}$). Check convergence by testing if $|z^3 - 1| < \epsilon$.

H3. Higher-Power Mandelbrot. Implement $z^n + c$ for general n. Use the polar form: if $z = re^{i\theta}$, then $z^n = r^n e^{in\theta}$. In code:

```
vec2 cpow(vec2 z, float n) {
    float r = length(z);
    float theta = atan(z.y, z.x);
    float rn = pow(r, n);
    return rn * vec2(cos(n * theta), sin(n * theta));
}
```

Try non-integer values of n! What happens at $n = 2.5$?

2.8.4. Projects

Project 1: Grid of Julia Sets

Create a grid where each cell shows the Julia set for a different value of c. The position in the grid determines c—effectively, each cell samples a point in the Mandelbrot parameter space.

When you zoom out, the overall pattern should reveal the Mandelbrot set: cells with connected Julia sets (solid regions) correspond to $c \in \mathcal{M}$, while cells with dust-like Julia sets correspond to $c \notin \mathcal{M}$.

Use the grid technique from Day 1:

```
float grid_size = 8.0;
vec2 cell_id = floor((p + 2.0) * grid_size / 4.0);
vec2 cell_p = fract((p + 2.0) * grid_size / 4.0) * 4.0 - 2.0;

// c comes from which cell we're in
vec2 c = (cell_id / grid_size) * 4.0 - 2.0;

// z starts from position within the cell
vec2 z = cell_p;
```

Project 2: Orbit Visualization

Instead of just coloring by iteration count, visualize the actual orbit of a point. Make the starting point draggable with the mouse:

- Let z_0 be the mouse position (normalized to the complex plane)
- Fix a parameter c (or let it be controllable too)
- Compute the first N iterates: $z_0, z_1, z_2, \ldots, z_N$
- Draw small circles at each iterate position
- Connect consecutive iterates with lines (use the segment SDF from Day 1!)
- Color by iteration index (early iterates one color, later iterates another)

This reveals the dynamics directly: bounded orbits stay in a region and may converge to a cycle, while escaping orbits spiral outward. Drag the starting point around and watch how the orbit changes—you'll see the sensitive dependence on initial conditions that makes chaos!

For Julia sets, fix `c` and drag `z_0`. For the Mandelbrot perspective, fix `z_0 = 0` and drag `c`.

3. Day 3: Tilings

Today we'll extend our per-pixel computations to study geometric tilings: we'll produce some beautiful hyperbolic tilings and if you work through the homework you'll be able to create something like below: a tiling morphs smoothly between two different views of the same geometry—the unbounded upper half-plane and the circular Poincaré disk.

Missing Demo

Shader demo `day3/hook-animated-tiling` not found.

Learning Objectives

By the end of today, you will be able to:

- Implement the folding algorithm to create tilings from any fundamental domain
- Use half-space abstractions to define arbitrary polygonal regions
- Understand why the algorithm works through the lens of reflection groups
- Apply the same techniques in hyperbolic geometry
- Convert between different models of hyperbolic space

3.1. The Folding Algorithm

Our goal today is to draw beautiful tilings—of the plane, the hyperbolic plane, and beyond—efficiently on the GPU. We'll learn a powerful technique: **fold any point back into a fundamental domain by repeatedly reflecting across boundaries**. The algorithm is simple, parallelizes perfectly, and works identically in Euclidean and hyperbolic geometry.

3.1.1. Tiling a Strip

Let's start with the simplest case: tiling the plane with horizontal strips. We define a **fundamental domain**—the strip where $0 < x < 1$—and reflect any point outside back in.

The reflection across a vertical line $x = c$ is simple: $(x, y) \mapsto (2c - x, y)$.

The algorithm:

- If $x < 0$, reflect across $x = 0$
- If $x > 1$, reflect across $x = 1$
- Repeat until the point stops moving

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Fold into the strip [0, 1]
    for (int i = 0; i < 20; i++) {
        if (p.x < 0.0) p.x = -p.x;           // Reflect across x = 0
        if (p.x > 1.0) p.x = 2.0 - p.x;      // Reflect across x = 1
    }

    // Draw a circle in the fundamental domain
    float d = length(p - vec2(0.5, 0.0));
    vec3 color = vec3(0.1, 0.1, 0.15);
    if (d < 0.3) {
        color = vec3(1.0, 0.8, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

Missing Demo

Shader demo `day3/strip-circle` not found.

The circle tiles the entire strip! But circles are symmetric—we can't tell if tiles are being reflected or just translated. Let's draw something asymmetric instead.

We'll use the letter "F", which has no mirror symmetry. We define a helper function that draws an F centered at the origin:

```
vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
    // Vertical bar
    if (p.x > -0.2 && p.x < -0.05 && p.y > -0.3 && p.y < 0.3) color = fgColor;
    // Top horizontal bar
    if (p.x > -0.2 && p.x < 0.2 && p.y > 0.15 && p.y < 0.3) color = fgColor;
    // Middle horizontal bar
    if (p.x > -0.2 && p.x < 0.1 && p.y > -0.05 && p.y < 0.1) color = fgColor;
    return color;
}
```

Then we replace the circle drawing with a call to `drawF`:

```
vec3 color = drawF(p - vec2(0.5, 0.0), vec3(0.1, 0.1, 0.15), vec3(1.0, 0.8, 0.3));
```

Missing Demo

Shader demo `day3/strip-F` not found.

Now we can see what's happening: the "F" alternates between normal and mirrored! This is reflection, not translation. Each time we cross a boundary, the image flips. We draw the shape once in the fundamental domain, and the folding algorithm tiles it everywhere.

3.1.2. Square Tiling

Extending to two dimensions is straightforward—just add boundaries for y:

```
// Fold into the square [0,1] × [0,1]
for (int i = 0; i < 20; i++) {
    if (p.x < 0.0) p.x = -p.x;
    if (p.x > 1.0) p.x = 2.0 - p.x;
    if (p.y < 0.0) p.y = -p.y;
    if (p.y > 1.0) p.y = 2.0 - p.y;
}
```

Missing Demo

Shader demo `day3/square-F` not found.

The "F" now tiles in both directions, with reflections across all four boundaries creating a kaleidoscopic pattern.

3.1.3. Counting Reflections

Let's track how many reflections were needed to reach the fundamental domain. This reveals the structure of the tiling:

```
// Fold into the square, counting reflections
int foldCount = 0;
for (int i = 0; i < 20; i++) {
    vec2 p0 = p;

    if (p.x < 0.0) { p.x = -p.x; foldCount++; }
    if (p.x > 1.0) { p.x = 2.0 - p.x; foldCount++; }
    if (p.y < 0.0) { p.y = -p.y; foldCount++; }
    if (p.y > 1.0) { p.y = 2.0 - p.y; foldCount++; }

    if (length(p - p0) < 0.0001) break;
}

// Color by fold count
```

```
float t = float(foldCount) / 10.0;
vec3 color = 0.5 + 0.5 * cos(6.28318 * (t + vec3(0.0, 0.33, 0.67)));
```

Missing Demo

Shader demo `day3/square-foldcount` not found.

Points near the fundamental domain (center of screen) need few reflections; points far away need many. The color bands show "distance" in terms of reflection count.

3.1.4. Parity Coloring

For tilings, we often want to distinguish adjacent tiles. The **parity** of the fold count (odd vs even) gives us a checkerboard pattern:

```
// After folding...
float parity = mod(float(foldCount), 2.0);
vec3 color = (parity < 0.5)
    ? vec3(0.9, 0.85, 0.8)   // Light
    : vec3(0.3, 0.35, 0.4);  // Dark
```

Missing Demo

Shader demo `day3/square-parity` not found.

This works because each reflection flips orientation—an odd number of reflections gives a mirror image of the fundamental domain, while an even number preserves orientation.

3.2. Half-Spaces and Reflections

Looking at our square tiling code, we see a repeated pattern: check if we're outside a boundary, reflect if so. Let's abstract this so we can handle arbitrary shapes—including triangles, which will be our gateway to hyperbolic geometry.

3.2.1. What is a Half-Space?

A **half-space** is one side of a line. Any line $ax + by = c$ divides the plane into two regions:

- Points where $ax + by < c$
- Points where $ax + by > c$

We encode a half-space by storing the line and which side we want:

```
struct HalfSpace {
    float a, b, c;  // Line: ax + by = c
    float side;     // +1 or -1: which side we want
};
```

The `side` parameter determines our inequality: we're "inside" the half-space when $(ax+by-c)\cdot\text{side} < 0$.

3.2.2. Inside Test and Reflection

Two functions do all the work. First, checking if a point is inside:

```
bool inside(vec2 p, HalfSpace h) {
    float val = h.a * p.x + h.b * p.y - h.c;
    return val * h.side < 0.0;
}
```

Second, reflecting into the half-space (only if we're outside):

```
vec2 reflectInto(vec2 p, HalfSpace h, inout int count) {
    float val = h.a * p.x + h.b * p.y - h.c;

    // Already inside?
    if (val * h.side < 0.0) return p;

    // Reflect across the boundary line
    vec2 n = vec2(h.a, h.b);
    n = n / length(n);  // Unit normal
    float dist = val / length(vec2(h.a, h.b));  // Signed distance to line
    count++;
    return p - 2.0 * dist * n;
}
```

> **GLSL: The `inout` Keyword**
>
> The `inout` keyword lets a function both read and modify a variable passed to it. When we write:
>
> ```
> vec2 reflectInto(vec2 p, HalfSpace h, inout int count)
> ```
>
> the `count` parameter is passed **by reference**—changes inside the function affect the original variable. This is how we track the total number of reflections: each call to `reflectInto` can increment the same counter.
> GLSL has three parameter modes: - `in` (default): pass by value, function gets a copy - `out`: function must write to it, caller receives the value - `inout`: function can read and write, changes persist
> This is similar to reference parameters in C++ or `ref` in C#.

The reflection formula comes from linear algebra: we move the point by twice its signed distance to the line, in the normal direction. We also increment the count each time we actually reflect.

3.2.3. Visualizing Half-Spaces

Let's see what a half-space looks like:

```
struct HalfSpace {
    float a, b, c;
    float side;
};

bool inside(vec2 p, HalfSpace h) {
    float val = h.a * p.x + h.b * p.y - h.c;
    return val * h.side < 0.0;
}
```

In `mainImage`, we test whether each point is inside and color accordingly:

```
// Half-space: x < 1 (left side of vertical line)
HalfSpace h = HalfSpace(1.0, 0.0, 1.0, 1.0);

vec3 color = inside(p, h) ? vec3(0.3, 0.5, 0.7) : vec3(0.1, 0.1, 0.15);

// Draw the boundary line
float dist = abs(h.a * p.x + h.b * p.y - h.c) / length(vec2(h.a, h.b));
if (dist < 0.03) color = vec3(1.0);
```

> Missing Demo
>
> Shader demo `day3/halfspace-single` not found.

The blue region is "inside" the half-space. Try changing the parameters to see different lines and sides!

3.2.4. Square Tiling with Half-Spaces

Let's rebuild our square tiling using this abstraction. The square $[0, 1] \times [0, 1]$ is defined by four half-spaces:

- Left edge ($x = 0$): want $x > 0$, so `HalfSpace(1.0, 0.0, 0.0, -1.0)`
- Right edge ($x = 1$): want $x < 1$, so `HalfSpace(1.0, 0.0, 1.0, 1.0)`
- Bottom edge ($y = 0$): want $y > 0$, so `HalfSpace(0.0, 1.0, 0.0, -1.0)`
- Top edge ($y = 1$): want $y < 1$, so `HalfSpace(0.0, 1.0, 1.0, 1.0)`

We need a function that reflects a point into a half-space if it's outside:

```
vec2 reflectInto(vec2 p, HalfSpace h, inout int count) {
    float val = h.a * p.x + h.b * p.y - h.c;
    if (val * h.side < 0.0) return p;  // Already inside

    vec2 n = vec2(h.a, h.b);
    n = n / length(n);
```

```
    float dist = val / length(vec2(h.a, h.b));
    count++;
    return p - 2.0 * dist * n;
}
```

Now the folding loop becomes:

```
// Four half-spaces defining [0,1] × [0,1]
HalfSpace left   = HalfSpace(1.0, 0.0, 0.0, -1.0);  // x > 0
HalfSpace right  = HalfSpace(1.0, 0.0, 1.0,  1.0);  // x < 1
HalfSpace bottom = HalfSpace(0.0, 1.0, 0.0, -1.0);  // y > 0
HalfSpace top    = HalfSpace(0.0, 1.0, 1.0,  1.0);  // y < 1

int foldCount = 0;
for (int i = 0; i < 20; i++) {
    vec2 p0 = p;
    p = reflectInto(p, left, foldCount);
    p = reflectInto(p, right, foldCount);
    p = reflectInto(p, bottom, foldCount);
    p = reflectInto(p, top, foldCount);
    if (length(p - p0) < 0.0001) break;
}
```

Missing Demo

Shader demo `day3/square-halfspace` not found.

It works! The result is identical to our earlier square tiling, but now the code is structured around half-spaces.

3.2.5. Triangle Tiling

Now the payoff—changing from a square to a triangle just means changing the half-space definitions. An equilateral triangle centered at the origin:

```
// Three half-spaces defining equilateral triangle
HalfSpace h1 = HalfSpace(0.0, 1.0, -0.5, -1.0);           // Bottom edge
HalfSpace h2 = HalfSpace(0.866, -0.5, -0.5, -1.0);        // Upper-right edge
HalfSpace h3 = HalfSpace(-0.866, -0.5, -0.5, -1.0);       // Upper-left edge

// Fold into triangle (three reflections instead of four)
for (int i = 0; i < 30; i++) {
    vec2 p0 = p;
    p = reflectInto(p, h1, foldCount);
    p = reflectInto(p, h2, foldCount);
    p = reflectInto(p, h3, foldCount);
    if (length(p - p0) < 0.0001) break;
}
```

Missing Demo

Shader demo `day3/triangle-tiling` not found.

Beautiful! The "F" shows us exactly how each triangle relates to its neighbors through reflection.

The Power of Abstraction

Compare the square and triangle tilings. The folding loop is identical—only the half-space definitions change. This abstraction will pay off enormously when we move to hyperbolic geometry: the algorithm stays the same, we just need different reflection operations!

3.3. Why It Works

We've seen the folding algorithm work for strips, squares, and triangles. But *why* does it work? The answer comes from **group theory**.

3.3.1. Reflection Groups

Each reflection across a half-space boundary is an **isometry**—a transformation that preserves distances. When we compose reflections, we get more isometries. The set of all such compositions forms a **group** called a **reflection group** (or Coxeter group).

For our triangle tiling, the group is generated by three reflections r_1, r_2, r_3 across the three edges. Every element of the group is a finite product of these generators:

$$g = r_{i_1} \circ r_{i_2} \circ \cdots \circ r_{i_k}$$

3.3.2. Fundamental Domains

Our triangle (or square) is a **fundamental domain** for the group action. This means:

1. The images of the fundamental domain under all group elements **tile the plane**: every point lies in some image $g(F)$
2. Different images **don't overlap** (except on boundaries): if $g \neq h$, then $g(F)$ and $h(F)$ have disjoint interiors

3.3.3. Why the Algorithm Terminates

[PROOF HERE]

3.3.4. The Orbit Map

What our shader computes is the **orbit map**: for each point p, find the unique group element g such that $g(p) \in F$. The folded position is $g(p)$, and the fold count tells us $|g|$ (the length of g as a word in the generators).

Parity coloring works because reflections have determinant -1: an odd number of reflections reverses orientation, an even number preserves it.

3.4. Hyperbolic Geometry

So far we've tiled the Euclidean plane. Now we'll tile the **hyperbolic plane**—a geometry with constant negative curvature, where the folding algorithm works exactly the same way.

3.4.1. The Upper Half-Plane Model

We represent the hyperbolic plane as the **upper half-plane**:

$$\mathbb{H}^2 = \{z = x + iy \in \mathbb{C} \,:\, y > 0\}$$

The real axis $y = 0$ is the **boundary at infinity**—not part of the space, but infinitely far away from every interior point.

What makes this hyperbolic rather than Euclidean is the **metric**: distances are scaled by $1/y$. Near the boundary (y small), distances are stretched enormously. High up (y large), distances are compressed. The Riemannian metric is:

$$ds^2 = \frac{dx^2 + dy^2}{y^2}$$

3.4.2. Hyperbolic Distance

The distance between two points $z_1 = x_1 + iy_1$ and $z_2 = x_2 + iy_2$ is:

$$d(z_1, z_2) = \operatorname{arccosh}\left(1 + \frac{|z_1 - z_2|^2}{2y_1 y_2}\right)$$

This formula captures the key property: distances grow without bound as either point approaches the boundary ($y \to 0$). The boundary is infinitely far from any interior point.

3.5. Hyperbolic Reflections and Tilings

3.5.1. Geodesics

In hyperbolic geometry, the "straight lines" are called **geodesics**. In the upper half-plane model, geodesics come in two types:

1. **Vertical lines**: $\{x = c\}$ for any constant c
2. **Semicircles**: centered on the real axis

Both meet the boundary at right angles.

3.5.2. Hyperbolic Half-Spaces

Just as in Euclidean geometry, a geodesic divides the plane into two **half-spaces**. We define a struct for each type:

```
struct HalfSpaceVert {
    float x;      // vertical line at x = c
    float side;   // +1: want x < c, -1: want x > c
};

struct HalfSpaceCirc {
    float center;   // center of semicircle (on real axis)
    float radius;   // radius of semicircle
    float side;     // +1: want outside circle, -1: want inside
};
```

Let's visualize one of each:

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
    float radius;
    float side;
};

bool inside(vec2 z, HalfSpaceVert h) {
    return (z.x - h.x) * h.side < 0.0;
}

bool inside(vec2 z, HalfSpaceCirc h) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);
    return (dist2 - h.radius * h.radius) * h.side > 0.0;
}
```

Missing Demo

Shader demo `day3/hyp-halfspaces` not found.

The blue region is "inside" the vertical half-space (where $x > 1$). The orange-tinted region is "inside" the circular half-space (outside the semicircle). Where they overlap, the colors blend.

3.5.3. Hyperbolic Reflections

Now we define `reflectInto` for each type. For a vertical half-space, we flip the x-coordinate:

```
vec2 reflectInto(vec2 z, HalfSpaceVert h, inout int count) {
    if ((z.x - h.x) * h.side < 0.0) return z;  // Already inside
    z.x = 2.0 * h.x - z.x;
    count++;
    return z;
}
```

For a circular half-space, we use **circle inversion**:

```
vec2 reflectInto(vec2 z, HalfSpaceCirc h, inout int count) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);

    if ((dist2 - h.radius * h.radius) * h.side > 0.0) return z;  // Already inside

    // Circle inversion
    z.x -= h.center;
    z /= h.radius;
    z /= dot(z, z);
    z *= h.radius;
    z.x += h.center;

    count++;
    return z;
}
```

> **Connection to Day 2**
>
> This is exactly the circle inversion formula from Day 2! When we inverted through circles in the Apollonian gasket, we were performing **hyperbolic reflections**. The Apollonian gasket lives in hyperbolic space—we just didn't know it yet.

Let's see these reflections in action. We'll place our "F" and reflect it across each type of half-space, alternating between them:

Missing Demo

Shader demo `day3/hyp-reflect-F` not found.

The shader alternates every 2 seconds: first reflecting across the vertical line (F flips horizontally), then across the semicircle (F gets inverted through the circle, distorting its shape).

3.5.4. The (2,3,∞) Triangle

Now we can tile the hyperbolic plane! The $(2, 3, \infty)$ triangle has angles $\pi/2$, $\pi/3$, and 0 (an ideal vertex at infinity). We set it up with:

- **Left boundary**: Vertical line at $x = 0$, want $x > 0$
- **Right boundary**: Vertical line at $x = 1/2$, want $x < 1/2$
- **Bottom boundary**: Unit semicircle centered at origin, want outside

The vertices are at i (angle $\pi/2$), at $\frac{1}{2} + \frac{\sqrt{3}}{2}i$ (angle $\pi/3$), and at infinity where the two vertical lines meet (angle 0).

Missing Demo

Shader demo `day3/hyp-triangle-halfspaces` not found.

The blue region is our fundamental domain—the $(2, 3, \infty)$ triangle where all three half-space conditions are satisfied. Now let's tile!

The `reflectInto` functions are the same as before, now with `inout int count` to track reflections. The `mainImage` is:

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 z = normalize_coord(fragCoord);

    // (2,3,∞) triangle
    HalfSpaceVert left = HalfSpaceVert(0.0, -1.0);        // x > 0
    HalfSpaceVert right = HalfSpaceVert(0.5, 1.0);        // x < 0.5
    HalfSpaceCirc bottom = HalfSpaceCirc(0.0, 1.0, 1.0);  // outside unit circle

    int foldCount = 0;
    for (int i = 0; i < 100; i++) {
        vec2 z0 = z;
        z = reflectInto(z, left, foldCount);
        z = reflectInto(z, right, foldCount);
        z = reflectInto(z, bottom, foldCount);
        if (length(z - z0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
    vec3 color = (parity < 0.5) ? vec3(0.85, 0.8, 0.75) : vec3(0.35, 0.4, 0.45);

    fragColor = vec4(color, 1.0);
}
```

Missing Demo

Shader demo `day3/hyp-tiling-23inf` not found.

The hyperbolic tiling emerges! Notice how the triangles appear to shrink as they approach the boundary—they're all the same hyperbolic size, but Euclidean distances compress near $y = 0$.

Compare to Euclidean

The structure is identical to our Euclidean tilings:
Euclidean triangle:

```
p = reflectInto(p, h1, foldCount);
p = reflectInto(p, h2, foldCount);
p = reflectInto(p, h3, foldCount);
```

Hyperbolic triangle:

```
z = reflectInto(z, left, foldCount);
z = reflectInto(z, right, foldCount);
z = reflectInto(z, bottom, foldCount);
```

Same loop structure, same counting. The only difference is the type of half-space (and thus which `reflectInto` overload is called). This is the power of abstraction!

3.6. Other Models

The upper half-plane is just one way to visualize hyperbolic geometry. The **Poincaré disk model** fits the entire hyperbolic plane inside a unit disk, making the global structure easier to see.

3.6.1. The Poincaré Disk

In the Poincaré disk model: - The hyperbolic plane is the open unit disk $\{z : |z| < 1\}$ - The boundary circle $|z| = 1$ represents infinity - Geodesics are circular arcs perpendicular to the boundary (and diameters)

3.6.2. The Cayley Transform

We convert between models using the **Cayley transform**:

$$w = \frac{z - i}{z + i}$$

This maps the upper half-plane to the unit disk, sending $i \mapsto 0$ and the real axis to the unit circle. The inverse is:

$$z = i\frac{1 + w}{1 - w}$$

To display our tiling in the Poincaré disk, we add complex arithmetic helpers and a conversion function:

```
// Complex multiplication and division
vec2 cmul(vec2 a, vec2 b) {
    return vec2(a.x * b.x - a.y * b.y, a.x * b.y + a.y * b.x);
}

vec2 cdiv(vec2 a, vec2 b) {
    float denom = dot(b, b);
    return vec2(a.x * b.x + a.y * b.y, a.y * b.x - a.x * b.y) / denom;
}

// Poincaré disk to upper half-plane
vec2 diskToUHP(vec2 w) {
    vec2 i = vec2(0.0, 1.0);
    vec2 one = vec2(1.0, 0.0);
    return cmul(i, cdiv(one + w, one - w));
}
```

In `mainImage`, we start with disk coordinates and convert to UHP before folding:

```
vec2 w = normalize_coord(fragCoord);  // Disk coordinates
vec2 z = diskToUHP(w);                // Convert to UHP

// ... same folding code as before ...

// Darken outside disk
if (length(w) > 1.0) color = vec3(0.05);
```

Missing Demo

Shader demo `day3/poincare-disk` not found.

The same tiling, now visible in its entirety! The disk model shows the beautiful self-similarity of hyperbolic tilings—triangles recede toward the boundary in an infinite cascade.

3.6.3. Hyperbolic F Tiling

Let's add our "F" to see the reflections clearly. We draw the F in the fundamental domain, and the folding algorithm automatically tiles it across the hyperbolic plane:

Missing Demo

Shader demo `day3/poincare-disk-F` not found.

The "F" tiles the hyperbolic plane! Notice how reflected copies flip orientation—each triangle's F is a mirror image of its neighbors, exactly as in our Euclidean tilings.

3.7. Exercises

3.7.1. Checkpoints

C1. Hyperbolic Circles. Draw hyperbolic circles around the mouse position. Use the distance formula:

$$d(z_1, z_2) = \operatorname{arccosh}\left(1 + \frac{|z_1 - z_2|^2}{2y_1 y_2}\right)$$

In GLSL, $\operatorname{arccosh}(x) = \log(x + \sqrt{x^2 - 1})$.

Draw two circles: one with hyperbolic radius 0.1 (small) and one with radius 0.5 (larger). Color pixels where the distance from the mouse is within 0.02 of these target radii.

Notice how the circles change shape as you move the mouse toward the boundary—they stretch horizontally and compress vertically, reflecting the hyperbolic metric.

C2. Edges of the (2,3,∞) Tiling. Add edge drawing to the hyperbolic tiling. After folding to the fundamental domain, check hyperbolic distance to each geodesic boundary.

Distance to vertical geodesic at $x = c$:

$$d(z, \text{geodesic}) = \operatorname{arccosh}\left(\frac{|z - c|}{y}\right)$$

where $|z - c| = \sqrt{(x - c)^2 + y^2}$ is the Euclidean distance to the point $(c, 0)$.

```
float distToGeodesic(vec2 z, HalfSpaceVert h) {
    z.x -= h.x;
    return acosh(length(z) / z.y);
}
```

Distance to semicircular geodesic centered at c with radius r:

The trick is to apply a Möbius transformation that sends the semicircle to a vertical line, then use the formula above. The transformation $w = \frac{z-(c+r)}{z-(c-r)}$ sends the endpoints $c \pm r$ to 0 and ∞, mapping the semicircle to the imaginary axis.

```
float distToGeodesic(vec2 z, HalfSpaceCirc h) {
    // Möbius transform sending semicircle to imaginary axis
    vec2 num = z - vec2(h.center + h.radius, 0.0);
    vec2 denom = z - vec2(h.center - h.radius, 0.0);
    vec2 w = cdiv(num, denom);

    // Distance to vertical geodesic at x = 0
    return acosh(length(w) / w.y);
}
```

Draw the edge white where distance < 0.03. The edges should have consistent hyperbolic thickness—appearing to fan out near the boundary in Euclidean terms.

3.7.2. Explorations

E1. Vertices and Edges in Both Models. Extend C2 to also draw vertices. The $(2, 3, \infty)$ triangle has vertices at:

- $z = i$ (angle $\pi/2$)
- $z = \frac{1}{2} + \frac{\sqrt{3}}{2}i$ (angle $\pi/3$)
- $z = \infty$ (ideal vertex—don't draw this one)

After folding, check if the folded point is near one of these vertices and draw a small circle.

Then adapt your shader to work in the Poincaré disk: apply your edge/vertex drawing after folding but before converting back. The same tiling should appear, now curved into the disk.

E2. The Klein Disk Model. The **Klein disk** (or Beltrami-Klein model) represents the hyperbolic plane in a unit disk where geodesics appear as *straight Euclidean line segments*. This makes some properties clearer but distorts angles.

The conversions are:

Poincaré → Klein:

$$k = \frac{2w}{1 + |w|^2}$$

Klein → Poincaré:

$$w = \frac{k}{1 + \sqrt{1 - |k|^2}}$$

Implement these and display the $(2, 3, \infty)$ tiling in the Klein model. Notice how the curved geodesics become straight lines! The triangles look like ordinary Euclidean triangles, but they're all the same hyperbolic size.

E3. The Band Model. The **band model** maps the hyperbolic plane to an infinite horizontal strip of height π. It's useful for visualizing hyperbolic translations.

The conversion from Poincaré disk uses the complex logarithm:

Poincaré → Band:

$$b = \log\left(\frac{1 + w}{1 - w}\right)$$

where the result has real part in $(-\infty, \infty)$ and imaginary part in $(0, \pi)$.

In GLSL, implement this as:

```
vec2 poincareToBand(vec2 w) {
    // (1+w)/(1-w) as complex division
    vec2 num = vec2(1.0, 0.0) + w;
    vec2 denom = vec2(1.0, 0.0) - w;
    vec2 ratio = cdiv(num, denom);
    // Complex log: log|z| + i*arg(z)
    return vec2(0.5 * log(dot(ratio, ratio)), atan(ratio.y, ratio.x));
}
```

Display the tiling in the band model. Horizontal translation in the band corresponds to hyperbolic translation along a geodesic!

E4. Interactive Möbius Transformations. The isometries of the hyperbolic plane are **Möbius transformations**:

$$z \mapsto \frac{az + b}{cz + d}$$

where a, b, c, d are real and $ad - bc = 1$ (for the upper half-plane).

These transformations preserve hyperbolic distances and map geodesics to geodesics. They form the group $\mathrm{PSL}(2, \mathbb{R})$.

Some useful isometries:

Horizontal translation by distance t:

$$z \mapsto z + t$$

Scaling (hyperbolic translation along the y-axis) by factor k:

$$z \mapsto kz$$

Rotation around i by angle θ:

$$z \mapsto \frac{z\cos(\theta/2) + \sin(\theta/2)}{-z\sin(\theta/2) + \cos(\theta/2)}$$

Make the tiling interactive: use mouse x-position to control horizontal translation and mouse y-position to control scaling. The tiling should slide and zoom while preserving its structure.

3.7.3. Challenges

H1. Animated Model Transitions. Create a smooth animation morphing between the upper half-plane and Poincaré disk.

The key insight: the Cayley transform $w = \frac{z-i}{z+i}$ is itself a Möbius transformation. You can interpolate between the identity and the Cayley transform using a one-parameter family.

One approach: the Cayley transform sends $i \mapsto 0$. Construct a family of Möbius transformations T_t where T_0 is the identity and T_1 is the Cayley transform. Use the matrix representation and interpolate.

Alternatively, interpolate the visual effect: blend between UHP coordinates and disk coordinates based on time.

H2. Dual Tiling. Every tiling has a **dual** obtained by connecting the centers of adjacent tiles.

For the triangle tiling, find the incenter (or centroid) of each fundamental triangle. Draw geodesics connecting centers of adjacent triangles instead of the original edges.

Hint: after folding, you know which fundamental domain you're in. The center of a $(2, 3, \infty)$ triangle is at approximately $z = 0.25 + 1.1i$. Use the fold sequence to transform this center to each tile.

H3. Single-Edge Tilings. Instead of drawing all three edges of each triangle, draw only ONE edge (say, the circular geodesic at the bottom).

This creates a different pattern: pairs of triangles glued along the missing edges form quadrilaterals. You get the "order-2" truncation of the original tiling.

Try drawing only the left vertical edge, or only the right vertical edge. Each choice produces a different pattern!

3.7.4. Projects

P1. General (p,q,r) Triangle Tilings. Implement tilings for arbitrary triangle groups (p, q, r) where the angles are π/p, π/q, and π/r.

The geometry: For a hyperbolic triangle (where $\frac{1}{p} + \frac{1}{q} + \frac{1}{r} < 1$), you need to compute the geodesics forming the edges. This requires solving for:

1. The positions of the three vertices
2. The semicircles (or vertical lines) connecting them

For a triangle with one ideal vertex (like $(2, 3, \infty)$), two edges are vertical lines and one is a semicircle. For a compact triangle (like $(2, 3, 7)$), all three edges are semicircles.

The algorithm: 1. Place one vertex at a convenient location (e.g., on the positive imaginary axis) 2. Use the angle constraints to determine the other vertices 3. Compute the semicircle through each pair of vertices

Implementation: - Allow the user to input p, q, r (perhaps via uniforms or compile-time constants) - Validate that the triangle is hyperbolic - Compute and draw the tiling with edges and vertices - Show the tiling in both UHP and Poincaré disk models

This is a substantial project requiring both geometric reasoning and careful implementation!

A. Appendix: Day 1 Shader Code

This appendix provides complete, standalone code for each shader referenced in Day 1. Each listing can be copied directly into Shadertoy and run immediately.

```
vec2 normalize_coord(vec2 coord) {
    vec2 uv = coord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);
    vec2 sun = normalize_coord(iMouse.xy);

    // Earth orbits the sun
    float orbit_radius = 0.8;
    vec2 earth = sun + orbit_radius * vec2(cos(iTime), sin(iTime));

    // Draw sun (larger, yellow)
    float d_sun = length(p - sun);
    // Draw earth (smaller, blue)
    float d_earth = length(p - earth);

    vec3 color = vec3(0.02, 0.02, 0.05);  // dark background
    if (d_sun < 0.3) {
        color = vec3(1.0, 0.9, 0.2);  // yellow sun
    }
    if (d_earth < 0.15) {
        color = vec3(0.2, 0.5, 1.0);  // blue earth
    }

    fragColor = vec4(color, 1.0);
}
```

A.1. A1. red

The simplest shader: every pixel is red.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    fragColor = vec4(1.0, 0.0, 0.0, 1.0);
}
```

A.2. A2. red-pulsing

Red channel oscillates with time.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    float red = 0.5 + 0.5 * sin(iTime);
    fragColor = vec4(red, 0.0, 0.0, 1.0);
}
```

A.3. A3. coordinates

Visualizing the coordinate system as color.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    fragColor = vec4(uv.x, uv.y, 0.0, 1.0);
}
```

A.4. A4. half-plane

Dividing the plane into two regions based on y-coordinate.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float L = p.y;
```

```
    vec3 color;
    if (L < 0.0) {
        color = vec3(1.0, 0.0, 0.0);  // red below
    } else {
        color = vec3(0.0, 0.0, 1.0);  // blue above
    }

    fragColor = vec4(color, 1.0);
}
```

A.5. A5. half-plane-animated

Animated line dividing the plane, with rotating normal and shifting offset.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float a = cos(iTime);
    float b = sin(iTime);
    float c = 0.5 * sin(iTime * 0.7);
    float L = a * p.x + b * p.y + c;

    vec3 color;
    if (L < 0.0) {
        color = vec3(1.0, 0.0, 0.0);
    } else {
        color = vec3(0.0, 0.0, 1.0);
    }

    fragColor = vec4(color, 1.0);
}
```

A.6. A6. circle

Filled circle centered at the origin.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float d = length(p);
    float r = 1.0;
    float f = d - r;

    vec3 color;
    if (f < 0.0) {
        color = vec3(1.0, 1.0, 0.0);  // yellow inside
    } else {
        color = vec3(0.1, 0.1, 0.3);  // dark blue outside
    }

    fragColor = vec4(color, 1.0);
}
```

A.7. A7. circle-curve

Circle outline (ring) centered at the origin.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float d = length(p);
    float r = 1.0;
    float eps = 0.1;
    float f = abs(d - r) - eps;

    vec3 color;
    if (f < 0.0) {
        color = vec3(1.0, 1.0, 0.0);  // yellow ring
    } else {
        color = vec3(0.1, 0.1, 0.3);  // dark background
    }

    fragColor = vec4(color, 1.0);
}
```

A.8. A8. parabola

The parabola $y = x^2$ rendered as an implicit curve.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float F = p.y - p.x * p.x;
    float eps = 0.1;

    vec3 color;
    if (abs(F) < eps) {
        color = vec3(1.0, 1.0, 0.0);  // yellow curve
    } else {
        color = vec3(0.1, 0.1, 0.3);  // dark background
    }

    fragColor = vec4(color, 1.0);
}
```

A.9. A9. lemniscate-naive

Lemniscate of Bernoulli with naive thresholding (non-uniform thickness).

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float a = 1.5;
    float r2 = dot(p, p);
    float F = r2 * r2 - a * a * (p.x * p.x - p.y * p.y);

    float eps = 0.15;

    vec3 color;
    if (abs(F) < eps) {
```

```
        color = vec3(1.0, 1.0, 0.0);
    } else {
        color = vec3(0.1, 0.1, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

A.10. A10. lemniscate-gradient

Lemniscate with gradient correction for uniform thickness.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float a = 1.5;
    float r2 = dot(p, p);
    float F = r2 * r2 - a * a * (p.x * p.x - p.y * p.y);

    vec2 grad = vec2(
        4.0 * p.x * r2 - 2.0 * a * a * p.x,
        4.0 * p.y * r2 + 2.0 * a * a * p.y
    );

    float dist = abs(F) / max(length(grad), 0.01);
    float eps = 0.05;

    vec3 color;
    if (dist < eps) {
        color = vec3(1.0, 1.0, 0.0);
    } else {
        color = vec3(0.1, 0.1, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

A.11. A11. lemniscate-animated

Cassini ovals animated through the lemniscate transition.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    // Cassini oval parameters
    float c = 1.0;  // half-distance between foci
    float a = 0.8 + 0.4 * sin(iTime * 0.5);  // animate through transition

    // Implicit equation: (x² + y²)² - 2c²(x² - y²) = a⁴ - c⁴
    float r2 = dot(p, p);
    float c2 = c * c;
    float a4 = a * a * a * a;
    float c4 = c2 * c2;
    float F = r2 * r2 - 2.0 * c2 * (p.x * p.x - p.y * p.y) - (a4 - c4);

    // Gradient
    vec2 grad = vec2(
        4.0 * p.x * r2 - 4.0 * c2 * p.x,
        4.0 * p.y * r2 + 4.0 * c2 * p.y
    );

    float dist = abs(F) / max(length(grad), 0.01);
    float eps = 0.05;

    vec3 color;
    if (dist < eps) {
        color = vec3(1.0, 1.0, 0.0);
    } else {
        color = vec3(0.1, 0.1, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

A.12. A12. circle-mouse

Circle that follows the mouse position.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    // Normalize fragment coordinate
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    // Normalize mouse coordinate the same way
    vec2 mouse = iMouse.xy / iResolution.xy;
    mouse = mouse - vec2(0.5, 0.5);
    mouse.x *= iResolution.x / iResolution.y;
    mouse = mouse * 4.0;

    // Circle centered at mouse
    float d = length(p - mouse);
    float r = 0.5;

    vec3 color;
    if (d < r) {
        color = vec3(1.0, 0.9, 0.2);
    } else {
        color = vec3(0.1, 0.1, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

A.13. A13. sun-earth

Sun at mouse click position with orbiting earth.

```
vec2 normalize_coord(vec2 coord) {
    vec2 uv = coord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Use iMouse.zw (last click position) so sun stays put
    vec2 sun = normalize_coord(iMouse.zw);
```

```
    // Earth orbits the sun
    float orbit_radius = 0.8;
    vec2 earth = sun + orbit_radius * vec2(cos(iTime), sin(iTime));

    // Draw sun (larger, yellow)
    float d_sun = length(p - sun);
    // Draw earth (smaller, blue)
    float d_earth = length(p - earth);

    vec3 color = vec3(0.02, 0.02, 0.05);  // dark background
    if (d_sun < 0.3) {
        color = vec3(1.0, 0.9, 0.2);  // yellow sun
    }
    if (d_earth < 0.15) {
        color = vec3(0.2, 0.5, 1.0);  // blue earth
    }

    fragColor = vec4(color, 1.0);
}
```

A.14. A14. folium-mouse

Folium of Descartes with mouse-controlled level set.

```
vec2 normalize_coord(vec2 coord) {
    vec2 uv = coord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Fixed parameter a
    float a = 1.5;

    // Map mouse x to level set value c in [-2, 2]
    float c = mix(-2.0, 2.0, iMouse.x / iResolution.x);

    // Folium of Descartes: x³ + y³ - 3axy = c
    float F = p.x*p.x*p.x + p.y*p.y*p.y - 3.0*a*p.x*p.y - c;

    // Gradient: ∇F = (3x² - 3ay, 3y² - 3ax)
    vec2 grad = vec2(3.0*p.x*p.x - 3.0*a*p.y, 3.0*p.y*p.y - 3.0*a*p.x);
```

```
    float dist = abs(F) / max(length(grad), 0.01);

    vec3 color;
    if (dist < 0.05) {
        color = vec3(1.0, 1.0, 0.0);
    } else {
        color = vec3(0.1, 0.1, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

A.15. A15. elliptic-family

Family of elliptic curves with mouse-controlled center parameters.

```
vec2 normalize_coord(vec2 coord) {
    vec2 uv = coord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Mouse controls the central (a, b) of our family
    float a_center = mix(-3.0, 1.0, iMouse.x / iResolution.x);
    float b_center = mix(-2.0, 2.0, iMouse.y / iResolution.y);

    vec3 color = vec3(0.05, 0.05, 0.1);  // dark background

    // Draw curves for a range of a values around a_center
    for (float i = -3.0; i ⩽ 3.0; i += 1.0) {
        float a = a_center + i * 0.5;  // more spacing
        float b = b_center;

        // Elliptic curve: y² = x³ + ax + b
        float F = p.y * p.y - p.x * p.x * p.x - a * p.x - b;

        // Gradient
        vec2 grad = vec2(-3.0 * p.x * p.x - a, 2.0 * p.y);
        float dist = abs(F) / max(length(grad), 0.01);

        // Brightness fades quickly: central curve bright, outer curves fade to background
```

```
        float t = abs(i) / 3.0;  // 0 at center, 1 at edges
        float brightness = 1.0 - t * t;  // quadratic falloff

        if (dist < 0.03 && brightness > 0.05) {
            // Check discriminant for this specific curve
            float disc = 4.0 * a * a * a + 27.0 * b * b;
            if (abs(disc) < 0.3) {
                color = mix(color, vec3(1.0, 0.3, 0.3), brightness);  // red for singular
            } else {
                color = mix(color, vec3(1.0, 1.0, 0.5), brightness);  // yellow for smooth
            }
        }
    }

    fragColor = vec4(color, 1.0);
}
```

A.16. A16. grid-circles

Grid of circles with square cells.

```
void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    vec2 p = uv * 4.0;

    float aspect = iResolution.x / iResolution.y;
    float N = 5.0;  // number of columns
    float L = (4.0 * aspect) / N;  // cell size

    vec2 cell_id = floor(p / L);
    vec2 cell_p = mod(p + vec2(L/2.0, L/2.0), L) - vec2(L/2.0, L/2.0);

    // Checkerboard background
    float checker = mod(cell_id.x + cell_id.y, 2.0);
    vec3 bg = mix(vec3(0.15, 0.15, 0.25), vec3(0.25, 0.15, 0.15), checker);

    // Circle in each cell
    float d = length(cell_p);
    float r = L * 0.35;

    vec3 color;
    if (d < r) {
        color = vec3(1.0, 1.0, 0.0);
```

```
    } else {
        color = bg;
    }

    fragColor = vec4(color, 1.0);
}
```

A.17. Notes

A.17.1. Coordinate Setup

Most shaders use this standard coordinate setup:

```
vec2 uv = fragCoord / iResolution.xy;   // normalize to [0,1]
uv = uv - vec2(0.5, 0.5);               // center origin
uv.x *= iResolution.x / iResolution.y;  // aspect correction
vec2 p = uv * 4.0;                      // scale to [-2, 2] range
```

A.17.2. Helper Function

For shaders using mouse input, we define:

```
vec2 normalize_coord(vec2 coord) {
    vec2 uv = coord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}
```

A.17.3. Gradient Correction

For implicit curves $F(x, y) = 0$ with uniform thickness:

```
float dist = abs(F) / max(length(grad), 0.01);
```

where `grad` is ∇F computed analytically.

B. Appendix: Day 2 Shader Code

Complete, standalone code for each shader referenced in Day 2. Each listing can be copied directly into Shadertoy and run immediately.

B.1. Common Functions

These helper functions are used throughout Day 2:

```
// Normalize screen coordinates to centered, aspect-corrected space
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

// Complex multiplication: (a + bi)(c + di)
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

// Squared magnitude of complex number (avoids sqrt)
float cabs2(vec2 z) {
    return dot(z, z);
}

// Cosine palette for smooth coloring
vec3 palette(float t) {
    vec3 a = vec3(0.5, 0.5, 0.5);
    vec3 b = vec3(0.5, 0.5, 0.5);
    vec3 c = vec3(1.0, 1.0, 1.0);
    vec3 d = vec3(0.00, 0.33, 0.67);
    return a + b * cos(6.28318 * (c * t + d));
}
```

B.2. A1. mandelbrot-zoom

Animated zoom into the Mandelbrot set with smooth coloring.

B. Appendix: Day 2 Shader Code

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

vec3 palette(float t) {
    vec3 a = vec3(0.5, 0.5, 0.5);
    vec3 b = vec3(0.5, 0.5, 0.5);
    vec3 c = vec3(1.0, 1.0, 1.0);
    vec3 d = vec3(0.00, 0.33, 0.67);
    return a + b * cos(6.28318 * (c * t + d));
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    float zoom = pow(1.5, mod(iTime, 30.0));
    return uv * 4.0 / zoom;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Zoom into the seahorse valley
    vec2 center = vec2(-0.745, 0.186);
    vec2 c = center + p;

    vec2 z = vec2(0.0, 0.0);
    int max_iter = 200;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (cabs2(z) > 4.0) break;
        z = cmul(z, z) + c;
    }

    vec3 color;
    if (iter == max_iter) {
        color = vec3(0.0);
    } else {
        // Smooth coloring
        float log_zn = log(cabs2(z)) / 2.0;
        float nu = log(log_zn / log(2.0)) / log(2.0);
        float smooth_iter = float(iter) + 1.0 - nu;
        float t = smooth_iter / float(max_iter);
```

```
        color = palette(t * 4.0);
    }

    fragColor = vec4(color, 1.0);
}
```

B.3. A2. mandelbrot-bw

Black and white Mandelbrot set.

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    vec2 c = p;
    c.x -= 0.5;  // shift left to center the interesting part

    vec2 z = vec2(0.0, 0.0);
    int max_iter = 100;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (cabs2(z) > 4.0) break;
        z = cmul(z, z) + c;
    }

    vec3 color;
    if (iter == max_iter) {
        color = vec3(0.0);
    } else {
        color = vec3(1.0);
    }
```

```
    fragColor = vec4(color, 1.0);
}
```

B.4. A3. mandelbrot-gray

Mandelbrot set with grayscale iteration coloring.

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    vec2 c = p;
    c.x -= 0.5;

    vec2 z = vec2(0.0, 0.0);
    int max_iter = 100;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (cabs2(z) > 4.0) break;
        z = cmul(z, z) + c;
    }

    vec3 color;
    if (iter == max_iter) {
        color = vec3(0.0);
    } else {
        float t = float(iter) / float(max_iter);
        color = vec3(t);
    }

    fragColor = vec4(color, 1.0);
}
```

B.5. A4. mandelbrot-color

Mandelbrot set with cosine palette coloring.

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

vec3 palette(float t) {
    vec3 a = vec3(0.5, 0.5, 0.5);
    vec3 b = vec3(0.5, 0.5, 0.5);
    vec3 c = vec3(1.0, 1.0, 1.0);
    vec3 d = vec3(0.00, 0.33, 0.67);
    return a + b * cos(6.28318 * (c * t + d));
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    vec2 c = p;
    c.x -= 0.5;

    vec2 z = vec2(0.0, 0.0);
    int max_iter = 100;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (cabs2(z) > 4.0) break;
        z = cmul(z, z) + c;
    }

    vec3 color;
    if (iter == max_iter) {
        color = vec3(0.0);
    } else {
        float t = float(iter) / float(max_iter);
        color = palette(t);
    }
```

```
    fragColor = vec4(color, 1.0);
}
```

B.6. A5. julia-static

Julia set with fixed parameter (black and white).

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Fixed parameter - try different values!
    vec2 c = vec2(-0.7, 0.27015);

    // z starts at pixel position
    vec2 z = p;

    int max_iter = 100;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (cabs2(z) > 4.0) break;
        z = cmul(z, z) + c;
    }

    vec3 color;
    if (iter == max_iter) {
        color = vec3(0.0);  // In the set: black
    } else {
        color = vec3(1.0);  // Escaped: white
    }

    fragColor = vec4(color, 1.0);
}
```

B.7. A6. julia-explorer

Interactive Julia set explorer: gray Mandelbrot as parameter space background, black Julia set overlaid, red dot shows current c.

```
vec2 cmul(vec2 z, vec2 w) {
    return vec2(z.x * w.x - z.y * w.y, z.x * w.y + z.y * w.x);
}

float cabs2(vec2 z) {
    return dot(z, z);
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Get c from mouse position
    vec2 c = normalize_coord(iMouse.xy);
    c.x -= 0.5;

    // Default to interesting value if no mouse
    if (iMouse.x < 1.0) {
        c = vec2(-0.7, 0.27015);
    }

    // Mandelbrot iteration (for background)
    vec2 mc = p;
    mc.x -= 0.5;
    vec2 mz = vec2(0.0);
    int m_iter;
    for (m_iter = 0; m_iter < 100; m_iter++) {
        if (cabs2(mz) > 4.0) break;
        mz = cmul(mz, mz) + mc;
    }

    // Julia iteration (for foreground)
    vec2 jz = p;
    int j_iter;
    for (j_iter = 0; j_iter < 100; j_iter++) {
        if (cabs2(jz) > 4.0) break;
        jz = cmul(jz, jz) + c;
    }
```

```
    // Color: light background, Mandelbrot in gray, Julia in black
    vec3 color = vec3(0.9);  // light background (escaped both)
    if (m_iter == 100) {
        color = vec3(0.6);  // Mandelbrot set in gray
    }
    if (j_iter == 100) {
        color = vec3(0.0);  // Julia set in black
    }

    // Draw red dot at c position (in Mandelbrot coordinates)
    vec2 c_pos = c;
    c_pos.x += 0.5;  // undo the offset we applied to c
    if (length(p - c_pos) < 0.05) {
        color = vec3(1.0, 0.0, 0.0);
    }

    fragColor = vec4(color, 1.0);
}
```

B.8. A7. inversion-toggle

Circle inversion visualization with toggling.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

vec2 invert(vec2 p) {
    return p / dot(p, p);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);
    vec2 p_inv = invert(p);

    // Toggle between original and inverted every second
    float time = fract(iTime * 0.5);
    vec2 q;
    if (time < 0.5) {
        q = p;
    } else {
        q = p_inv;
    }

    vec3 color = vec3(0.1, 0.1, 0.15);
```

```
    // Draw the unit circle
    float d_unit = abs(length(p) - 1.0);
    if (d_unit < 0.02) color = vec3(0.5, 0.5, 0.5);

    // Draw a vertical line at x = 2
    if (abs(q.x - 2.0) < 0.02) color = vec3(1.0, 1.0, 0.0);

    // Draw a horizontal line at y = 1.5
    if (abs(q.y - 1.5) < 0.02) color = vec3(1.0, 1.0, 0.0);

    // Draw a circle centered at (2, 0) with radius 0.5
    float d_circle = abs(length(q - vec2(2.0, 0.0)) - 0.5);
    if (d_circle < 0.02) color = vec3(1.0, 1.0, 0.0);

    fragColor = vec4(color, 1.0);
}
```

B.9. A8. inversion-grid

Circle inversion of a grid.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

vec2 invert(vec2 p) {
    return p / dot(p, p);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);
    vec2 p_inv = invert(p);

    // Toggle
    float time = fract(iTime * 0.5);
    vec2 q;
    if (time < 0.5) {
        q = p;
    } else {
        q = p_inv;
    }

    vec3 color = vec3(0.1, 0.1, 0.15);
```

```
    // Draw the unit circle
    float d_unit = abs(length(p) - 1.0);
    if (d_unit < 0.02) color = vec3(0.5, 0.5, 0.5);

    // Draw a grid using mod
    vec2 grid = mod(q, 0.5);
    if (grid.x < 0.02 || grid.y < 0.02) color = vec3(1.0, 1.0, 0.0);

    fragColor = vec4(color, 1.0);
}
```

B.10. A9. inversion-moving

Inversion through a moving circle.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

struct Circle {
    vec2 center;
    float radius;
};

vec2 invert(vec2 p, Circle c) {
    vec2 d = p - c.center;
    return c.center + c.radius * c.radius * d / dot(d, d);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Animate the inversion circle
    Circle inv_circle;
    inv_circle.center = vec2(sin(iTime) * 0.5, cos(iTime * 0.7) * 0.5);
    inv_circle.radius = 1.0 + 0.3 * sin(iTime * 1.3);

    vec2 p_inv = invert(p, inv_circle);

    vec3 color = vec3(0.1, 0.1, 0.15);

    // Draw the inversion circle
    float d_inv = abs(length(p - inv_circle.center) - inv_circle.radius);
    if (d_inv < 0.02) color = vec3(0.5, 0.5, 0.5);
```

```
    // Draw a grid in the inverted space
    vec2 grid = mod(p_inv, 0.5);
    if (grid.x < 0.02 || grid.y < 0.02) color = vec3(1.0, 1.0, 0.0);

    fragColor = vec4(color, 1.0);
}
```

B.11. A10. apollonian-setup

The four mutually tangent circles (no iteration).

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 6.0;
}

struct Circle {
    vec2 center;
    float radius;
};

float distToCircle(vec2 p, Circle c) {
    return abs(length(p - c.center) - c.radius);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Three mutually tangent inner circles plus outer circle
    // For three circles of radius r centered at vertices of equilateral triangle:
    // - Side length of triangle = 2r (so circles touch)
    // - Circumradius of triangle = 2r / sqrt(3)
    // - Outer circle radius = circumradius + r

    float r = 1.0;
    float triSide = 2.0 * r;
    float circumradius = triSide / sqrt(3.0);

    // Inner circles at vertices of equilateral triangle
    Circle c1 = Circle(vec2(0.0, circumradius), r);
    Circle c2 = Circle(vec2(-circumradius * 0.866, -circumradius * 0.5), r);
    Circle c3 = Circle(vec2(circumradius * 0.866, -circumradius * 0.5), r);

    // Outer circle tangent to all three from outside
    Circle outer = Circle(vec2(0.0, 0.0), circumradius + r);
```

```
    vec3 color = vec3(0.1, 0.1, 0.15);

    // Draw all four circles
    if (distToCircle(p, c1) < 0.03) color = vec3(1.0, 0.3, 0.3);
    if (distToCircle(p, c2) < 0.03) color = vec3(0.3, 1.0, 0.3);
    if (distToCircle(p, c3) < 0.03) color = vec3(0.3, 0.3, 1.0);
    if (distToCircle(p, outer) < 0.03) color = vec3(1.0, 1.0, 1.0);

    fragColor = vec4(color, 1.0);
}
```

B.12. A11. apollonian-iterated

Full Apollonian gasket with iteration.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 6.0;
}

struct Circle {
    vec2 center;
    float radius;
};

vec2 invert(vec2 p, Circle c) {
    vec2 d = p - c.center;
    return c.center + c.radius * c.radius * d / dot(d, d);
}

float distToCircle(vec2 p, Circle c) {
    return abs(length(p - c.center) - c.radius);
}

bool isInside(vec2 p, Circle c) {
    return length(p - c.center) < c.radius;
}

vec3 palette(float t) {
    vec3 a = vec3(0.5, 0.5, 0.5);
    vec3 b = vec3(0.5, 0.5, 0.5);
    vec3 c = vec3(1.0, 1.0, 1.0);
    vec3 d = vec3(0.00, 0.33, 0.67);
    return a + b * cos(6.28318 * (c * t + d));
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
```

```
{
    vec2 p = normalize_coord(fragCoord);

    // Setup circles with correct geometry
    float r = 1.0;
    float triSide = 2.0 * r;
    float circumradius = triSide / sqrt(3.0);

    Circle c1 = Circle(vec2(0.0, circumradius), r);
    Circle c2 = Circle(vec2(-circumradius * 0.866, -circumradius * 0.5), r);
    Circle c3 = Circle(vec2(circumradius * 0.866, -circumradius * 0.5), r);
    Circle outer = Circle(vec2(0.0, 0.0), circumradius + r);

    // Iterate inversions
    int max_iter = 50;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (isInside(p, c1)) {
            p = invert(p, c1);
        } else if (isInside(p, c2)) {
            p = invert(p, c2);
        } else if (isInside(p, c3)) {
            p = invert(p, c3);
        } else if (!isInside(p, outer)) {
            p = invert(p, outer);
        } else {
            break;
        }
    }

    // Color by iteration count
    float t = float(iter) / float(max_iter);
    vec3 color = palette(t);

    // Draw circle boundaries
    float dMin = min(min(distToCircle(p, c1), distToCircle(p, c2)),
                     min(distToCircle(p, c3), distToCircle(p, outer)));
    if (dMin < 0.02) color = vec3(1.0);

    fragColor = vec4(color, 1.0);
}
```

B.13. A12. apollonian-final

Apollonian gasket with coloring emphasizing the limit set.

B. Appendix: Day 2 Shader Code

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 6.0;
}

struct Circle {
    vec2 center;
    float radius;
};

vec2 invert(vec2 p, Circle c) {
    vec2 d = p - c.center;
    return c.center + c.radius * c.radius * d / dot(d, d);
}

bool isInside(vec2 p, Circle c) {
    return length(p - c.center) < c.radius;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Setup circles with correct geometry
    float r = 1.0;
    float triSide = 2.0 * r;
    float circumradius = triSide / sqrt(3.0);

    Circle c1 = Circle(vec2(0.0, circumradius), r);
    Circle c2 = Circle(vec2(-circumradius * 0.866, -circumradius * 0.5), r);
    Circle c3 = Circle(vec2(circumradius * 0.866, -circumradius * 0.5), r);
    Circle outer = Circle(vec2(0.0, 0.0), circumradius + r);

    // Iterate inversions
    int max_iter = 100;
    int iter;

    for (iter = 0; iter < max_iter; iter++) {
        if (isInside(p, c1)) {
            p = invert(p, c1);
        } else if (isInside(p, c2)) {
            p = invert(p, c2);
        } else if (isInside(p, c3)) {
            p = invert(p, c3);
        } else if (!isInside(p, outer)) {
            p = invert(p, outer);
        } else {
            break;
```

```
        }
    }

    // Color by iteration count, emphasizing the limit set
    float t = float(iter) / float(max_iter);
    vec3 color = 30.0 * vec3(pow(t, 2.0));

    fragColor = vec4(color, 1.0);
}
```

C. Day 3: Complete Shader Code

This document provides complete, standalone shader code for each demo in Day 3. Copy any of these directly into Shadertoy to run.

C.0.1. strip-circle

Basic strip tiling with a circle in the fundamental domain.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    // Fold into the strip [0, 1]
    for (int i = 0; i < 20; i++) {
        if (p.x < 0.0) p.x = -p.x;
        if (p.x > 1.0) p.x = 2.0 - p.x;
    }

    // Draw a circle in the fundamental domain
    float d = length(p - vec2(0.5, 0.0));
    vec3 color = vec3(0.1, 0.1, 0.15);
    if (d < 0.3) {
        color = vec3(1.0, 0.8, 0.3);
    }

    fragColor = vec4(color, 1.0);
}
```

C.0.2. strip-F

Strip tiling with the letter F to show reflection behavior.

```
vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
    if (p.x > -0.2 && p.x < -0.05 && p.y > -0.3 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.2 && p.y > 0.15 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.1 && p.y > -0.05 && p.y < 0.1) color = fgColor;
    return color;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    for (int i = 0; i < 20; i++) {
        if (p.x < 0.0) p.x = -p.x;
        if (p.x > 1.0) p.x = 2.0 - p.x;
    }

    vec3 color = drawF(p - vec2(0.5, 0.0), vec3(0.1, 0.1, 0.15), vec3(1.0, 0.8, 0.3));
    fragColor = vec4(color, 1.0);
}
```

C.0.3. square-F

Square tiling with the letter F.

```
vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
    if (p.x > -0.2 && p.x < -0.05 && p.y > -0.3 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.2 && p.y > 0.15 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.1 && p.y > -0.05 && p.y < 0.1) color = fgColor;
    return color;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
```

```
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    for (int i = 0; i < 20; i++) {
        if (p.x < 0.0) p.x = -p.x;
        if (p.x > 1.0) p.x = 2.0 - p.x;
        if (p.y < 0.0) p.y = -p.y;
        if (p.y > 1.0) p.y = 2.0 - p.y;
    }

    vec3 color = drawF(p - vec2(0.5, 0.5), vec3(0.1, 0.1, 0.15), vec3(1.0, 0.8, 0.3));
    fragColor = vec4(color, 1.0);
}
```

C.0.4. square-foldcount

Square tiling colored by number of reflections.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    int foldCount = 0;
    for (int i = 0; i < 20; i++) {
        vec2 p0 = p;
        if (p.x < 0.0) { p.x = -p.x; foldCount++; }
        if (p.x > 1.0) { p.x = 2.0 - p.x; foldCount++; }
        if (p.y < 0.0) { p.y = -p.y; foldCount++; }
        if (p.y > 1.0) { p.y = 2.0 - p.y; foldCount++; }
        if (length(p - p0) < 0.0001) break;
    }

    float t = float(foldCount) / 10.0;
    vec3 color = 0.5 + 0.5 * cos(6.28318 * (t + vec3(0.0, 0.33, 0.67)));
```

```
    fragColor = vec4(color, 1.0);
}
```

C.0.5. square-parity

Square tiling with checkerboard parity coloring.

```
vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    int foldCount = 0;
    for (int i = 0; i < 20; i++) {
        vec2 p0 = p;
        if (p.x < 0.0) { p.x = -p.x; foldCount++; }
        if (p.x > 1.0) { p.x = 2.0 - p.x; foldCount++; }
        if (p.y < 0.0) { p.y = -p.y; foldCount++; }
        if (p.y > 1.0) { p.y = 2.0 - p.y; foldCount++; }
        if (length(p - p0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
    vec3 color = (parity < 0.5) ? vec3(0.9, 0.85, 0.8) : vec3(0.3, 0.35, 0.4);
    fragColor = vec4(color, 1.0);
}
```

C.0.6. halfspace-single

Visualization of a single half-space.

```
struct HalfSpace {
    float a, b, c;
    float side;
};

bool inside(vec2 p, HalfSpace h) {
```

```
    float val = h.a * p.x + h.b * p.y - h.c;
    return val * h.side < 0.0;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 4.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    HalfSpace h = HalfSpace(1.0, 0.0, 1.0, 1.0);

    vec3 color = inside(p, h) ? vec3(0.3, 0.5, 0.7) : vec3(0.1, 0.1, 0.15);

    float dist = abs(h.a * p.x + h.b * p.y - h.c) / length(vec2(h.a, h.b));
    if (dist < 0.03) color = vec3(1.0);

    fragColor = vec4(color, 1.0);
}
```

C.0.7. square-halfspace

Square tiling using half-space abstraction with F and parity coloring.

```
struct HalfSpace {
    float a, b, c;
    float side;
};

vec2 reflectInto(vec2 p, HalfSpace h, inout int count) {
    float val = h.a * p.x + h.b * p.y - h.c;
    if (val * h.side < 0.0) return p;

    vec2 n = vec2(h.a, h.b);
    n = n / length(n);
    float dist = val / length(vec2(h.a, h.b));
    count++;
    return p - 2.0 * dist * n;
}

vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
```

```
    if (p.x > -0.2 && p.x < -0.05 && p.y > -0.3 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.2 && p.y > 0.15 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.1 && p.y > -0.05 && p.y < 0.1) color = fgColor;
    return color;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 8.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    HalfSpace left   = HalfSpace(1.0, 0.0, 0.0, -1.0);
    HalfSpace right  = HalfSpace(1.0, 0.0, 1.0,  1.0);
    HalfSpace bottom = HalfSpace(0.0, 1.0, 0.0, -1.0);
    HalfSpace top    = HalfSpace(0.0, 1.0, 1.0,  1.0);

    int foldCount = 0;
    for (int i = 0; i < 20; i++) {
        vec2 p0 = p;
        p = reflectInto(p, left, foldCount);
        p = reflectInto(p, right, foldCount);
        p = reflectInto(p, bottom, foldCount);
        p = reflectInto(p, top, foldCount);
        if (length(p - p0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
    vec3 bg = (parity < 0.5) ? vec3(0.85, 0.8, 0.75) : vec3(0.35, 0.4, 0.45);
    vec3 fg = (parity < 0.5) ? vec3(0.6, 0.2, 0.2) : vec3(0.2, 0.2, 0.6);

    vec3 color = drawF(p - vec2(0.5, 0.5), bg, fg);
    fragColor = vec4(color, 1.0);
}
```

C.0.8. triangle-tiling

Euclidean equilateral triangle tiling.

```
struct HalfSpace {
    float a, b, c;
    float side;
```

```
};

vec2 reflectInto(vec2 p, HalfSpace h, inout int count) {
    float val = h.a * p.x + h.b * p.y - h.c;
    if (val * h.side < 0.0) return p;

    vec2 n = vec2(h.a, h.b);
    n = n / length(n);
    float dist = val / length(vec2(h.a, h.b));
    count++;
    return p - 2.0 * dist * n;
}

vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
    if (p.x > -0.15 && p.x < 0.0 && p.y > -0.2 && p.y < 0.2) color = fgColor;
    if (p.x > -0.15 && p.x < 0.15 && p.y > 0.1 && p.y < 0.2) color = fgColor;
    if (p.x > -0.15 && p.x < 0.08 && p.y > -0.02 && p.y < 0.08) color = fgColor;
    return color;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.5);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 6.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 p = normalize_coord(fragCoord);

    HalfSpace h1 = HalfSpace(0.0, 1.0, -0.5, -1.0);
    HalfSpace h2 = HalfSpace(0.866, -0.5, -0.5, -1.0);
    HalfSpace h3 = HalfSpace(-0.866, -0.5, -0.5, -1.0);

    int foldCount = 0;
    for (int i = 0; i < 30; i++) {
        vec2 p0 = p;
        p = reflectInto(p, h1, foldCount);
        p = reflectInto(p, h2, foldCount);
        p = reflectInto(p, h3, foldCount);
        if (length(p - p0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
    vec3 bg = (parity < 0.5) ? vec3(0.85, 0.8, 0.75) : vec3(0.35, 0.4, 0.45);
    vec3 fg = (parity < 0.5) ? vec3(0.6, 0.2, 0.2) : vec3(0.2, 0.2, 0.6);

    vec3 color = drawF(p, bg, fg);
```

```
    fragColor = vec4(color, 1.0);
}
```

C.0.9. hyp-halfspaces

Hyperbolic half-spaces in the upper half-plane.

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
    float radius;
    float side;
};

bool inside(vec2 z, HalfSpaceVert h) {
    return (z.x - h.x) * h.side < 0.0;
}

bool inside(vec2 z, HalfSpaceCirc h) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);
    return (dist2 - h.radius * h.radius) * h.side > 0.0;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv.y *= iResolution.y / iResolution.x;
    return uv * 5.0 - vec2(1.0, 0.0);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 z = normalize_coord(fragCoord);

    HalfSpaceVert hv = HalfSpaceVert(1.0, -1.0);
    HalfSpaceCirc hc = HalfSpaceCirc(2.5, 1.0, 1.0);

    vec3 color = vec3(0.1, 0.1, 0.15);

    if (inside(z, hv)) color = vec3(0.3, 0.5, 0.7);
    if (inside(z, hc)) color += vec3(0.4, 0.2, 0.1);

    if (abs(z.x - hv.x) < 0.03) color = vec3(1.0);
```

```
    if (abs(length(z - vec2(hc.center, 0.0)) - hc.radius) < 0.03 && z.y > 0.0) color = vec3(1.0);

    if (z.y < 0.02) color = vec3(0.15);

    fragColor = vec4(color, 1.0);
}
```

C.0.10. hyp-reflect-F

Hyperbolic reflections of F across vertical and circular geodesics.

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
    float radius;
    float side;
};

vec2 reflectInto(vec2 z, HalfSpaceVert h) {
    if ((z.x - h.x) * h.side < 0.0) return z;
    z.x = 2.0 * h.x - z.x;
    return z;
}

vec2 reflectInto(vec2 z, HalfSpaceCirc h) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);

    if ((dist2 - h.radius * h.radius) * h.side > 0.0) return z;

    z.x -= h.center;
    z /= h.radius;
    z /= dot(z, z);
    z *= h.radius;
    z.x += h.center;

    return z;
}

vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
    if (p.x > -0.2 && p.x < -0.05 && p.y > -0.3 && p.y < 0.3) color = fgColor;
    if (p.x > -0.2 && p.x < 0.2 && p.y > 0.15 && p.y < 0.3) color = fgColor;
```

```
    if (p.x > -0.2 && p.x < 0.1 && p.y > -0.05 && p.y < 0.1) color = fgColor;
    return color;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = uv - vec2(0.5, 0.0);
    uv.x *= iResolution.x / iResolution.y;
    return uv * 6.0;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 z = normalize_coord(fragCoord);

    HalfSpaceVert hv = HalfSpaceVert(-1.5, -1.0);
    HalfSpaceCirc hc = HalfSpaceCirc(1.5, 2.5, 1.0);

    float t = mod(iTime, 4.0);
    if (t < 2.0) {
        z = reflectInto(z, hv);
    } else {
        z = reflectInto(z, hc);
    }

    vec3 color = drawF(z - vec2(0.5, 3.5), vec3(0.1, 0.1, 0.15), vec3(1.0, 0.8, 0.3));

    vec2 z_orig = normalize_coord(fragCoord);

    if (abs(z_orig.x - hv.x) < 0.04) color = vec3(0.5);
    if (abs(length(z_orig - vec2(hc.center, 0.0)) - hc.radius) < 0.04 && z_orig.y > 0.0) color = ve
    if (z_orig.y < 0.02) color = vec3(0.15);

    fragColor = vec4(color, 1.0);
}
```

C.0.11. hyp-triangle-halfspaces

The three half-spaces defining the (2,3,∞) triangle.

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
```

```
    float radius;
    float side;
};

bool inside(vec2 z, HalfSpaceVert h) {
    return (z.x - h.x) * h.side < 0.0;
}

bool inside(vec2 z, HalfSpaceCirc h) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);
    return (dist2 - h.radius * h.radius) * h.side > 0.0;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv.y *= iResolution.y / iResolution.x;
    return uv * 4.0 - vec2(0.75, 0.0);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 z = normalize_coord(fragCoord);

    HalfSpaceVert left = HalfSpaceVert(0.0, -1.0);
    HalfSpaceVert right = HalfSpaceVert(0.5, 1.0);
    HalfSpaceCirc bottom = HalfSpaceCirc(0.0, 1.0, 1.0);

    vec3 color = vec3(0.1, 0.1, 0.15);
    if (inside(z, left) && inside(z, right) && inside(z, bottom)) {
        color = vec3(0.3, 0.5, 0.7);
    }

    if (abs(z.x - 0.0) < 0.02 && z.y > 0.0) color = vec3(1.0);
    if (abs(z.x - 0.5) < 0.02 && z.y > 0.0) color = vec3(1.0);
    if (abs(length(z) - 1.0) < 0.02 && z.y > 0.0) color = vec3(1.0);

    if (z.y < 0.01) color = vec3(0.15);

    fragColor = vec4(color, 1.0);
}
```

C.0.12. hyp-tiling-23inf

The (2,3,∞) hyperbolic triangle tiling in the upper half-plane.

C. Day 3: Complete Shader Code

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
    float radius;
    float side;
};

vec2 reflectInto(vec2 z, HalfSpaceVert h, inout int count) {
    float val = z.x - h.x;
    if (val * h.side < 0.0) return z;
    z.x = 2.0 * h.x - z.x;
    count++;
    return z;
}

vec2 reflectInto(vec2 z, HalfSpaceCirc h, inout int count) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);

    if ((dist2 - h.radius * h.radius) * h.side > 0.0) return z;

    z.x -= h.center;
    z /= h.radius;
    z /= dot(z, z);
    z *= h.radius;
    z.x += h.center;

    count++;
    return z;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv.y *= iResolution.y / iResolution.x;
    return uv * 4.0 - vec2(0.75, 0.0);
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 z = normalize_coord(fragCoord);

    HalfSpaceVert left = HalfSpaceVert(0.0, -1.0);
    HalfSpaceVert right = HalfSpaceVert(0.5, 1.0);
    HalfSpaceCirc bottom = HalfSpaceCirc(0.0, 1.0, 1.0);

    int foldCount = 0;
```

```
    for (int i = 0; i < 100; i++) {
        vec2 z0 = z;
        z = reflectInto(z, left, foldCount);
        z = reflectInto(z, right, foldCount);
        z = reflectInto(z, bottom, foldCount);
        if (length(z - z0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
    vec3 color = (parity < 0.5) ? vec3(0.85, 0.8, 0.75) : vec3(0.35, 0.4, 0.45);

    vec2 z_orig = normalize_coord(fragCoord);
    if (z_orig.y < 0.01) color = vec3(0.15);

    fragColor = vec4(color, 1.0);
}
```

C.0.13. poincare-disk

The (2,3,∞) tiling in the Poincaré disk model.

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
    float radius;
    float side;
};

vec2 reflectInto(vec2 z, HalfSpaceVert h, inout int count) {
    float val = z.x - h.x;
    if (val * h.side < 0.0) return z;
    z.x = 2.0 * h.x - z.x;
    count++;
    return z;
}

vec2 reflectInto(vec2 z, HalfSpaceCirc h, inout int count) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);

    if ((dist2 - h.radius * h.radius) * h.side > 0.0) return z;

    z.x -= h.center;
```

```
    z /= h.radius;
    z /= dot(z, z);
    z *= h.radius;
    z.x += h.center;

    count++;
    return z;
}

vec2 cmul(vec2 a, vec2 b) {
    return vec2(a.x * b.x - a.y * b.y, a.x * b.y + a.y * b.x);
}

vec2 cdiv(vec2 a, vec2 b) {
    float denom = dot(b, b);
    return vec2(a.x * b.x + a.y * b.y, a.y * b.x - a.x * b.y) / denom;
}

vec2 diskToUHP(vec2 w) {
    vec2 i = vec2(0.0, 1.0);
    vec2 one = vec2(1.0, 0.0);
    return cmul(i, cdiv(one + w, one - w));
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = (uv - 0.5) * 2.5;
    uv.x *= iResolution.x / iResolution.y;
    return uv;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 w = normalize_coord(fragCoord);
    vec2 z = diskToUHP(w);

    HalfSpaceVert left = HalfSpaceVert(0.0, -1.0);
    HalfSpaceVert right = HalfSpaceVert(0.5, 1.0);
    HalfSpaceCirc bottom = HalfSpaceCirc(0.0, 1.0, 1.0);

    int foldCount = 0;
    for (int i = 0; i < 100; i++) {
        vec2 z0 = z;
        z = reflectInto(z, left, foldCount);
        z = reflectInto(z, right, foldCount);
        z = reflectInto(z, bottom, foldCount);
        if (length(z - z0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
```

```
    vec3 color = (parity < 0.5) ? vec3(0.85, 0.8, 0.75) : vec3(0.35, 0.4, 0.45);

    if (length(w) > 1.0) color = vec3(0.05);

    fragColor = vec4(color, 1.0);
}
```

C.0.14. poincare-disk-F

The (2,3,∞) tiling in the Poincaré disk with F markers.

```
struct HalfSpaceVert {
    float x;
    float side;
};

struct HalfSpaceCirc {
    float center;
    float radius;
    float side;
};

vec2 reflectInto(vec2 z, HalfSpaceVert h, inout int count) {
    float val = z.x - h.x;
    if (val * h.side < 0.0) return z;
    z.x = 2.0 * h.x - z.x;
    count++;
    return z;
}

vec2 reflectInto(vec2 z, HalfSpaceCirc h, inout int count) {
    vec2 rel = z - vec2(h.center, 0.0);
    float dist2 = dot(rel, rel);

    if ((dist2 - h.radius * h.radius) * h.side > 0.0) return z;

    z.x -= h.center;
    z /= h.radius;
    z /= dot(z, z);
    z *= h.radius;
    z.x += h.center;

    count++;
    return z;
}

vec2 cmul(vec2 a, vec2 b) {
```

C. Day 3: Complete Shader Code

```
    return vec2(a.x * b.x - a.y * b.y, a.x * b.y + a.y * b.x);
}

vec2 cdiv(vec2 a, vec2 b) {
    float denom = dot(b, b);
    return vec2(a.x * b.x + a.y * b.y, a.y * b.x - a.x * b.y) / denom;
}

vec2 diskToUHP(vec2 w) {
    vec2 i = vec2(0.0, 1.0);
    vec2 one = vec2(1.0, 0.0);
    return cmul(i, cdiv(one + w, one - w));
}

vec3 drawF(vec2 p, vec3 bgColor, vec3 fgColor) {
    vec3 color = bgColor;
    if (p.x > -0.06 && p.x < -0.02 && p.y > -0.08 && p.y < 0.08) color = fgColor;
    if (p.x > -0.06 && p.x < 0.06 && p.y > 0.04 && p.y < 0.08) color = fgColor;
    if (p.x > -0.06 && p.x < 0.03 && p.y > -0.01 && p.y < 0.03) color = fgColor;
    return color;
}

vec2 normalize_coord(vec2 fragCoord) {
    vec2 uv = fragCoord / iResolution.xy;
    uv = (uv - 0.5) * 2.5;
    uv.x *= iResolution.x / iResolution.y;
    return uv;
}

void mainImage(out vec4 fragColor, in vec2 fragCoord)
{
    vec2 w = normalize_coord(fragCoord);
    vec2 z = diskToUHP(w);

    HalfSpaceVert left = HalfSpaceVert(0.0, -1.0);
    HalfSpaceVert right = HalfSpaceVert(0.5, 1.0);
    HalfSpaceCirc bottom = HalfSpaceCirc(0.0, 1.0, 1.0);

    int foldCount = 0;
    for (int i = 0; i < 100; i++) {
        vec2 z0 = z;
        z = reflectInto(z, left, foldCount);
        z = reflectInto(z, right, foldCount);
        z = reflectInto(z, bottom, foldCount);
        if (length(z - z0) < 0.0001) break;
    }

    float parity = mod(float(foldCount), 2.0);
    vec3 bg = (parity < 0.5) ? vec3(0.85, 0.8, 0.75) : vec3(0.35, 0.4, 0.45);
    vec3 fg = (parity < 0.5) ? vec3(0.6, 0.2, 0.2) : vec3(0.2, 0.2, 0.6);
```

```
    vec3 color = drawF(z - vec2(0.25, 1.2), bg, fg);

    if (length(w) > 1.0) color = vec3(0.05);

    fragColor = vec4(color, 1.0);
}
```